```
∸-irrelevant : ∀ {n m} → (p₁ p₂ : m ≤ n) → (n ∸ m) p₁ ≡ (n ∸ m) p₂
∸-irrelevant {n} {m} p₁ p₂ =
  cong (λ p → (n ∸ m) p) (≤-irrelevant p₁ p₂)

-- n ∸ m ≤ n
∸→≤ : ∀ {n m} → (m≤n : m ≤ n) → (n ∸ m) m≤n ≤ n
∸→≤ z≤n = ≤-refl
∸→≤ (s≤s m≤n) = ≤-trans (∸→≤ m≤n) n≤suc-n

n∸n≡0 : ∀ {n} → (n ∸ n) (≤-refl {n}) ≡ 0
n∸n≡0 {zero} = refl
n∸n≡0 {suc n} = n∸n≡0 {n}

-- suc (n ∸ m) ≡ suc n ∸ m
∸-suc : ∀ {n m} → {m≤n : m ≤ n}
        → suc ((n ∸ m) m≤n) ≡ (suc n ∸ m) (≤-trans m≤n n≤suc-n)
∸-suc {_} {zero} {z≤n} = refl
∸-suc {suc n} {suc m} {s≤s m≤n} = ∸-suc {n} {m} {m≤n}

n∸[n∸m]≡m : ∀ {m n} → (m≤n : m ≤ n)
            → (n ∸ ((n ∸ m) m≤n)) (∸→≤ m≤n) ≡ m
n∸[n∸m]≡m {zero} {n} z≤n = n∸n≡0 {n}
n∸[n∸m]≡m {suc m} {suc n} (s≤s m≤n) =
  trans (sym (∸-suc {n} {(n ∸ m) m≤n}))
        (cong suc (n∸[n∸m]≡m {m} {n} m≤n))

n+m∸m≡n : ∀ {m n} → (n + m ∸ m) (+→≤ʳ) ≡ n
n+m∸m≡n {m} {zero} =
  trans (∸-irrelevant {m} {m} +→≤ʳ ≤-refl) (n∸n≡0 {m})
n+m∸m≡n {m} {suc n} =
  trans
    (∸-irrelevant {suc n + m} {m} +→≤ʳ (≤-trans +→≤ʳ n≤suc-n))
    (trans (sym (∸-suc {n + m} {m}))
           (cong suc (n+m∸m≡n {m} {n})))

-- m ≤ n → m ∸ p ≤ n ∸ p
sub-monoʳ-≤ : ∀ {p m n} → (p≤m : p ≤ m) → (m≤n : m ≤ n)
              → (m ∸ p) p≤m ≤ (n ∸ p) (≤-trans p≤m m≤n)
sub-monoʳ-≤ z≤n m≤n = m≤n
sub-monoʳ-≤ (s≤s p≤m) (s≤s m≤n) = sub-monoʳ-≤ p≤m m≤n

-- suc d ≤ d' → d ≤ d' ∸ (d' ∸ (suc d))
suc-d≤d'→d≤d'∸[d'∸[suc-d]] : ∀ {d d'} → (δ₁≤δ₂ : suc d ≤ d')
        → d ≤ ((d' ∸ ((d' ∸ (suc d)) δ₁≤δ₂)) (∸→≤ δ₁≤δ₂))
suc-d≤d'→d≤d'∸[d'∸[suc-d]] δ₁≤δ₂ =
        m≡n,p≤n→p≤m (n∸[n∸m]≡m δ₁≤δ₂) n≤suc-n
```