Jack Gao

Using functor categories to generate intermediate code with Agda

Computer Science Tripos - Part II Dissertation

Homerton College

May 13, 2025

Declaration of Originality

I, the candidate for Part II of the Computer Science Tripos with Blind Grading Number 2330G, hereby declare that this report and the work described in it are my own work, unaided except as may be specified below, and that the report does not contain material that has already been used to any substantial extent for a comparable purpose. In preparation of this report, I adhered to the Department of Computer Science and Technology AI Policy. I am content for my report to be made available to the students and staff of the University.

Date: May 13, 2025

Proforma

Candidate Number:	2330G
Title of Project:	Using functor categories to generate intermediate code with Agda
Examination	Computer Science Tripos - Part II - 2025
Word-count:	[wordcount] [1]
Code line count:	[linecount] [2]
Project Originator:	Yulong Huang
Project Supervisor:	Yulong Huang and Yufeng Li

Original Aims of the Project

Work Completed

Special Difficulties

[1] This word-count was computed by `texcount -1 -sum -merge -q dissertation.tex`.
[2] This code line count was computed by `find . -name "*.agda" | xargs wc -l`.

Contents

1 Introduction **1**

1.1 Motivation . . . 2

1.2 Language choice: Agda's advantages . . . 2

1.3 Contributions . . . 3

2 Preparation **4**

2.1 Starting Point . . . 5

2.2 Category theory . . . 5

2.2.1 Category . . . 5

2.2.2 Isomorphism . . . 7

2.2.3 Cartesian closed category . . . 8

2.2.4 Functor . . . 10

2.2.5 Natural transformation . . . 10

2.2.6 Functor category . . . 11

2.2.7 Presheaf category . . . 11

2.2.8 Yoneda lemma . . . 11

2.2.9 Cartesian closed structure in presheaf categories . . . 12

2.3 Agda . . . 12

2.3.1 Basic data types and pattern matching . . . 12

2.3.2 Dependent Types . . . 13

2.3.3 Curry-Howard-Lambek correspondence . . . 13

2.3.4 Equality, congruence and substitution . . . 14

2.3.5 Standard library . . . 15

2.3.6 Interactive programming with holes . . . 16

2.4 Requirement Analysis . . . 16

3 Implementation **17**

3.1 Tools used . . . 18

3.2 Repository Overview . . . 18
3.2.1 Directory structure . . . 18
3.2.2 File dependencies and descriptions . . . 18
3.3 Source Language . . . 19
3.3.1 Types . . . 19
3.3.2 Contexts, variables and the lookup judgement . . . 20
3.3.3 Terms and the typing judgement . . . 21
3.3.4 Operational Semantics . . . 22
3.4 Target Language . . . 22
3.4.1 Stack descriptor . . . 23
3.4.2 Grammar . . . 25
3.5 Customised Library . . . 27
3.5.1 Definition from standard library . . . 27
3.5.2 Subtraction . . . 29
3.5.3 More properties of natural numbers . . . 32
3.6 Compiler . . . 33
3.6.1 Presheaf semantics . . . 33
3.6.2 Continuations . . . 34
3.6.3 Denotational semantics of types and contexts . . . 35
3.6.4 Functorial mapping . . . 35
3.6.5 Denotational semantics of terms . . . 37
3.6.6 Compilation . . . 41

4 Evaluation **42**
4.1 Tests . . . 42
4.1.1 Test cases . . . 42
4.1.2 Feature checklist . . . 43
4.2 Extensibility . . . 45
4.3 Success criteria . . . 45

5 Conclusion **46**
5.1 Results . . . 46
5.2 Lessons learned . . . 46
5.3 Future work . . . 47

A Operational Semantics of the source language **50**

Chapter 1

Introduction

Contents

1.1 **Motivation** 2

1.2 **Language choice: Agda's advantages** 2

1.3 **Contributions** 3

Programming languages act as a bridge between human thoughts and machine execution. They exist in two complementary realms: the human realm, where intent is expressed through abstractions like variables and functions, and the machine realm, where low-level instructions are executed on hardware.

Denotational semantics is related to the human realms. It provides a theoretical framework for defining the meaning of programming languages by interpreting them into mathematical objects. By formalising the denotational semantics of a programming language, we ensure that its abstraction align with our intentions.

Compilers are related to the machine realms. They are programs that translate high-level code into executable instructions, resolving abstractions into concrete operations like memory allocation and register management. Compilers are essential for turning human-written code into physical computation.

This project explores how the two process can be deeply connected by demonstrating how denotational semantics can directly generate a compiler. Simply typed lambda calculus (STLC) [1] is a well-studied programming language that serves as a foundation for many modern languages. It has denotational semantics in cartesian closed categories (CCC) [2]. As a CCC, presheaf categories over store locations can be used to model STLC with stores, as shown by by Reynolds [3] and Oles [4, 5]. Later, Reynolds presented a denotational semantics of STLC with stores in the form of a presheaf category over compiler states [6]. By interpreting the source language into the presheaf category over *stack descriptors*, where objects of the category represent instruction sequences parametrised by stack layouts, the semantic model directly yields a compiler.

In this dissertation, I implement Reynolds' presheaf-based compiler for STLC with stores in a dependently typed programming language, Agda [7]. The compiler is a *functor* (structure-preserving map) from the source language to the target language. The implementation is verified with Agda, which ensures that the input source program, the compilation process, and the output target program are all well-typed.

This work both validates and refines Reynolds' theory, offering a concrete example of how category-theoretic semantics can generate intermediate code. The project also serves as a practical demonstration of the power of dependently typed programming languages can mechanise the link between theory and practice.

1.1 Motivation

This project is motivated and guided by the following:

Motivation I. Formal verification of the compiler

Reynolds' work presented detailed definitions of denotational semantics of STLC with stores, which are complicated and error-prone. The rise of verified compilers including CompCert [8] and CakeML [9] reflects a broader trend toward trustworthy systems, where correctness proofs replace testing for critical guarantees. I aim to provide a formalisation of the definition in a proof assistant to verify the correctness of the given denotational semantics.

Motivation II. Implementation of the compiler

Reynolds concluded that he did not have a proper dependently typed programming language in hand, so his compiler remained a partial function [6, Ch.6]. I aim to provide a computer implementation of this theoretical framework in a dependently typed programming language.

1.2 Language choice: Agda's advantages

Agda [7] is a dependently typed programming language and proof assistant. Agda captures the source language's intrinsic syntax with indexed families, which contains only well-typed terms. Therefore, it focuses on the correct programs and rules out the ill-typed nonsensical inputs.

Dependently typed languages provide a natural framework for expressing functor categories is proven both theoretically and practically. There have been dependent-type-theoretic model of categories [10], and it has been shown that functor categories arise naturally as dependent function types [11]. A formalisation of Category Theory, including cartesian closed categories, functors and presheaves has been developed in Agda by Hu and Caratte [12]. Other proof assistants, such as Isabelle/Hol, does not have a dependently typed language structure, and thus cannot express the functor categories as naturally as Agda.

Compared to other dependently typed languages, Agda is more balanced in terms of programming and proving. Its with-abstraction and rewrite construction allow for a more flexible and powerful way to define and manipulate terms. The with-abstraction allows us to inspect intermediate values in a term, which gives a refined view of a function's argument. The rewrite construction allows us to define new terms by expanding existing terms, which avoids rewriting proofs of similar structures.

Agda also provides an interactive environment for writing and verifying programs, which will be further discussed in §2.3.6.

1.3 Contributions

Addressed the two motivations presented in §1.1 and contributed to the following:

Motivation I. Formal verification of the compiler

I formalised the terms in the source language and target language in Agda. I also refined some type definitions in the denotational semantics of the source language.

Motivation II. Implementation of the compiler

I implemented the compiler from the source language to the target language in Agda.

Chapter 2

Preparation

Contents

2.1 Starting Point **5**
2.2 Category theory **5**
2.2.1 Category 5
2.2.2 Isomorphism 7
2.2.3 Cartesian closed category 8
2.2.4 Functor 10
2.2.5 Natural transformation 10
2.2.6 Functor category 11
2.2.7 Presheaf category 11
2.2.8 Yoneda lemma 11
2.2.9 Cartesian closed structure in presheaf categories 12
2.3 Agda **12**
2.3.1 Basic data types and pattern matching 12
2.3.2 Dependent Types 13
2.3.3 Curry-Howard-Lambek correspondence 13
2.3.4 Equality, congruence and substitution 14
2.3.5 Standard library 15
2.3.6 Interactive programming with holes 16
2.4 Requirement Analysis **16**

For implementing the compiler, I need to understand the theoretical background of presheaves and functor categories that are used as the denotational semantics of the source language, and I need to understand Agda's dependent types to correctly express the dependent function space of presheaf exponentials. This chapter begins with my starting point and an introduction to category theory, followed by a brief overview of the dependently typed programming language Agda, and requirements analysis of the compiler.

2.1 Starting Point

Prior to this project, I had no experience with Agda. Although I was aware of the open-source online tutorial *Programming Language Foundations in Agda* (PLFA) [13], my preparation was limited to setting up the Agda environment on my laptop by following the "Front Matter" section of the tutorial.

I did not have any other experience with compiler beyond Part IB Compiler Construction Course. I had no prior exposure to category theory and type theory before the Part II lectures.

2.2 Category theory

Category theory provides a high-level abstraction from which we can reason about the structure of mathematical objects and their relationships. It provides us a "bird's eye view" of the mathematics which enable us to spot patterns that are difficult to see in the details [14]. More specifically, it provides a "purer" view of functions that is not derived from sets [15]. Compared to set theory which is "element-oriented", category theory is "function-oriented" and "morphism-oriented". We understand structures not via elements but by how they transform into each other.

Notation in category theory is similar to that in set theory and type theory. For example, $A, B, \ldots$ are used to denote objects, sets, or types, and arrows are used to denote morphisms or functions (e.g. $A \to B$). There is a deeper connection between category theory and type theory, which will be discussed in §2.3.3.

The following is a brief introduction to the basic concepts of category theory, which is based on the work of Leinster [14], Riehl [16], and the lecture notes of Part II Category Theory by Andrew Pitts and Marcelo Fiore [17].

2.2.1 Category

Definition 2.1 (Category). A *category* $\mathcal{C}$ is specified by

- a collection of objects $\mathbf{obj}(\mathcal{C})$, whose elements are called $\mathcal{C}$-objects;
- for each $X, Y \in \mathbf{obj}(\mathcal{C})$, a collection of morphisms $\mathcal{C}(X, Y)$, whose elements are called $\mathcal{C}$-morphisms from X to Y (e.g. $f : X \to Y$);
- for each $X \in \mathbf{obj}(\mathcal{C})$, an element $\mathbf{id}_X \in \mathcal{C}(X, X)$ called the identity morphism on X;
- for each $X, Y, Z \in \mathbf{obj}(\mathcal{C})$, a function

$$\begin{aligned} \mathcal{C}(X,Y) \times \mathcal{C}(Y,Z) &\to \mathcal{C}(X,Z) \\ (f,g) &\mapsto g \circ f \end{aligned}$$

  called the composition of morphisms;

satisfying the following properties:

- **(Unit Laws)** For all $X, Y \in \mathbf{obj}(\mathcal{C})$ and $f \in \mathcal{C}(X, Y)$, we have

$$\mathbf{id}_Y \circ f = f = f \circ \mathbf{id}_X \tag{2.1}$$

- **(Associativity Law)** For all $X, Y, Z, W \in \mathbf{obj}(\mathcal{C})$ and $f \in \mathcal{C}(X, Y)$, $g \in \mathcal{C}(Y, Z)$, $h \in \mathcal{C}(Z, W)$, we have

$$h \circ (g \circ f) = (h \circ g) \circ f \tag{2.2}$$

Example 2.2 ($\mathcal{S}et$). An example of category is the category of sets, denoted as $\mathcal{S}et$, specified by the following:

- The objects of $\mathcal{S}et$ are a fixed universe of sets;
- For each $X, Y \in \mathbf{obj}(\mathcal{S}et)$, the morphisms from $X \to Y$ in $\mathcal{S}et$ are the functions $X \to Y$;
- The identity morphism on X in $\mathcal{S}et$ is the identity function on X;
- The composition of morphisms in $\mathcal{S}et$ is defined as the composition of functions.

Associativity law and unit laws are satisfied in $\mathcal{S}et$, since the composition of functions is associative and the identity function compose with any function is the function itself.

Definition 2.3 (Preorder). A *preorder* $\underline{P} = (P, \sqsubseteq)$ is a set P with a binary relation $\sqsubseteq$ that is

- reflexive: $\forall x \in P, x \sqsubseteq x$;
- transitive: $\forall x, y, z \in P, x \sqsubseteq y \wedge y \sqsubseteq z \implies x \sqsubseteq z$.

Example 2.4 (Category determined by preorder). Another example of category is a category $\mathcal{C}_{\underline{P}}$ determined by any preorder $\underline{P}$, which is specified by the following:

- The objects of $\mathcal{C}_{\underline{P}}$ are the elements of P;
- $\mathcal{C}_{\underline{P}}(x, y) = \begin{cases} \{(x, y)\} & \text{if } x \sqsubseteq y \\ \emptyset & \text{otherwise} \end{cases}$
- The identity morphism on X in $\mathcal{C}_{\underline{P}}$ is the identity function on X;
- The composition of morphisms in $\mathcal{C}_{\underline{P}}$ is defined as the composition of functions.

Associativity law and unit laws are satisfied in $\mathcal{C}_{\underline{P}}$.

$\mathcal{C}_{\underline{P}}$ is used later for stack descriptors in the denotational semantics of the source language.

Opposite category

The idea of opposite category is that if we have a category $\mathcal{C}$, we can reverse the direction of all morphisms in $\mathcal{C}$ to obtain a new category $\mathcal{C}^{\mathbf{op}}$.

Definition 2.5 (Opposite category). Given a category $\mathcal{C}$, its *opposite category* $\mathcal{C}^{\mathbf{op}}$ is specified by the following:

- The objects of $\mathcal{C}^{\mathbf{op}}$ are the same as those of $\mathcal{C}$;
- For each $X, Y \in \mathbf{obj}(\mathcal{C})$, the morphisms from X to Y in $\mathcal{C}^{\mathbf{op}}$ are the morphisms from Y to X in $\mathcal{C}$;
- The identity morphism on X in $\mathcal{C}^{\mathbf{op}}$ is the identity morphism on X in $\mathcal{C}$;
- The composition of morphisms in $\mathcal{C}^{\mathbf{op}}$ is defined as the composition of morphisms in $\mathcal{C}$.

The notation of commutative diagram is widely used in category theory as a convenient visual representation of the relationships between objects and morphisms in a category.

Commutative diagrams

A *diagram* in a category $\mathcal{C}$ is a directed graph whose vertices are $\mathcal{C}$-objects and whose edges are $\mathcal{C}$-morphisms.

A diagram is *commutative* (or *commutes*) if any two finite paths in the graph between any two vertices X and Y in the diagram determine the equal morphism $f \in \mathcal{C}(X, Y)$ under the composition of morphisms.



Figure 2.1: Commutative diagram for Unit Laws



Figure 2.2: Commutative diagram for Associativity Law

As examples of commutative diagrams, Figure 2.2 and Figure 2.1 are commutative diagrams for the unit laws and associativity law respectively.

2.2.2 Isomorphism

Definition 2.6 (Isomorphism). Given a category $\mathcal{C}$, a C-morphism $f : X \to Y$ is called an *isomorphism* if there exists a C-morphism $g : Y \to X$ such that the following diagram com-

mutes:



(2.3)

In other words, f is an isomorphism if there exists a morphism g such that $g \circ f = \mathbf{id}_X$ and $f \circ g = \mathbf{id}_Y$.

The morphism g is uniquely determined by f and is called the *inverse* of f, denoted as f^{-1}.

Given two objects X and Y in a category $\mathcal{C}$, if there exists an isomorphism from X to Y, we say that X and Y are *isomorphic* in $\mathcal{C}$ and write $X \cong Y$.

2.2.3 Cartesian closed category

Terminal object

Definition 2.7 (Terminal object). Given a category $\mathcal{C}$, an object $T \in \mathbf{obj}(\mathcal{C})$ is called a *terminal object* if for all $X \in \mathbf{obj}(\mathcal{C})$, there exists a unique **C**-morphism $f : X \to T$.

Terminal objects are unique up to isomorphism. In other words, we have the following properties:

- If T and T' are both terminal objects in $\mathcal{C}$, then there exists a unique isomorphism $f : T \to T'$.
- If T is a terminal object in $\mathcal{C}$ and $T \cong T'$, then T' is also a terminal object in $\mathcal{C}$.

Example 2.8 (Terminal object in $\mathcal{S}et$). $\mathcal{S}et$ has a terminal object $\{*\}$, which is an arbitrary singleton set containing a single element $*$.

For any set X, there exists a unique function $f : X \to \{*\}$ that maps every element of X to the single element $*$ in $\{*\}$.

There is a unique isomorphism $f : \{*\} \to \{\cdot\}$ for any two singleton sets $\{*\}$ and $\{\cdot\}$, which is $f(*) = \cdot$.

Binary product

Definition 2.9 (Binary product). Given a category $\mathcal{C}$, the *binary product* of two objects X and Y in $\mathcal{C}$ is specified by

- a $\mathcal{C}$-object $X \times Y$;
- two $\mathcal{C}$-morphisms $\pi_1 : X \times Y \to X$ and $\pi_2 : X \times Y \to Y$ called the *projections* of $X \times Y$;

such that for all $Z \in \mathbf{obj}(\mathcal{C})$ and morphisms $f : Z \to X$ and $g : Z \to Y$, there exists a unique morphism $u : Z \to X \times Y$ such that the following diagram commutes in $\mathcal{C}$:



(2.4)

The unique morphism u is written as $\langle f, g \rangle : Z \to X \times Y$ where $f = \pi_1 \circ u$ and $g = \pi_2 \circ u$.

It can be shown that the binary product is unique up to (unique) isomorphism.

Example 2.10 (Binary product in $\mathcal{S}et$). The binary product of two sets X and Y in $\mathcal{S}et$ is the Cartesian product $X \times Y = \{(x, y) \mid x \in X \wedge y \in Y\}$, where (x, y) are ordered pairs.

We have the following projections:

- $\pi_1 : X \times Y \to X$ is defined as $\pi_1(x, y) = x$ for all $(x, y) \in X \times Y$;
- $\pi_2 : X \times Y \to Y$ is defined as $\pi_2(x, y) = y$ for all $(x, y) \in X \times Y$.

For any set Z, for any functions $f : Z \to X$ and $g : Z \to Y$, the unique morphism $u : Z \to X \times Y$ is defined as:

$$u(z) = (f(z), g(z)) \text{ for all } z \in Z.$$

Exponential

Definition 2.11 (Exponential). Given a category $\mathcal{C}$ with binary products, the *exponential* of two objects X and Y in $\mathcal{C}$ is specified by

- a $\mathcal{C}$-object $X \Rightarrow Y$;
- a $\mathcal{C}$-morphism $\mathbf{app} : (X \Rightarrow Y) \times X \to Y$ called the *application* of $X \Rightarrow Y$;

such that for all $Z \in \mathbf{obj}(\mathcal{C})$ and morphisms $f : Z \times X \to Y$, there exists a unique morphism $u : Z \to X \Rightarrow Y$ such that the following diagram commutes in $\mathcal{C}$:



(2.5)

We write $\mathbf{cur} f$ for the unique morphism u such that $f = \mathbf{app} \circ (\mathbf{cur} f \times \mathbf{id}_X)$, where $\mathbf{cur} f$ is called the *currying* of f.

It can be shown that the exponential is unique up to (unique) isomorphism.

Example 2.12 (Exponential in $\mathcal{S}et$). The exponential of two sets X and Y in $\mathcal{S}et$ is the set of all functions from X to Y.

Function application gives the morphism $\mathbf{app} : (X \Rightarrow Y) \times X \to Y$ as $\mathbf{app}(f, x) = f(x)$ for all $f \in X \Rightarrow Y$ and $x \in X$.

The currying operation transform a function $f : Z \times X \to Y$ into a function $\mathbf{cur}f : Z \to (X \Rightarrow Y)$, which is defined as $\mathbf{cur}f(z) = \lambda x.f(z, x)$ for all $z \in Z$ and $x \in X$.

Definition 2.13 (Cartesian closed category). A category $\mathcal{C}$ is called a *Cartesian closed category* (CCC) if it has a terminal object, binary products and exponentials of any two objects.

2.2.4 Functor

Definition 2.14 (Functor). Given two categories $\mathcal{C}$ and $\mathcal{D}$, a *functor* $F : \mathcal{C} \to \mathcal{D}$ is specified by

- a function referred to as object mapping

$$\begin{aligned} \mathbf{obj}(\mathcal{C}) &\to \mathbf{obj}(\mathcal{D}) \\ X &\mapsto F(X) \end{aligned}$$

- for each $X, Y \in \mathbf{obj}(\mathcal{C})$, a function referred to as functorial mapping

$$\begin{aligned} \mathcal{C}(X, Y) &\to \mathcal{D}(F(X), F(Y)) \\ f &\mapsto F(f) \end{aligned}$$

satisfying the following properties:

- For all $X, Y \in \mathbf{obj}(\mathcal{C})$ and $f \in \mathcal{C}(X, Y)$, we have: $F(\mathbf{id}_X) = \mathbf{id}_{F(X)}$
- For all $X, Y, Z \in \mathbf{obj}(\mathcal{C})$ and $f \in \mathcal{C}(X, Y)$, $g \in \mathcal{C}(Y, Z)$, we have: $F(g \circ f) = F(g) \circ F(f)$

2.2.5 Natural transformation

Definition 2.15 (Natural transformation). Given two categories $\mathcal{C}$ and $\mathcal{D}$, and two functors $F, G : \mathcal{C} \to \mathcal{D}$, a *natural transformation* $\theta : F \to G$ is a family of morphisms $\theta_X \in \mathcal{D}(F(X), G(X))$ for each $X \in \mathbf{obj}(\mathcal{C})$ such that for all $X, Y \in \mathbf{obj}(\mathcal{C})$ and $f \in \mathcal{C}(X, Y)$, the following diagram

$$\begin{array}{ccc} F(X) & \xrightarrow{\theta_X} & G(X) \\ {\scriptstyle F(f)}\downarrow & & \downarrow{\scriptstyle G(f)} \\ F(Y) & \xrightarrow[\theta_Y]{} & G(Y) \end{array} \tag{2.6}$$

commutes in $\mathcal{D}$, i.e. $G(f) \circ \theta_X = \theta_Y \circ F(f)$

2.2.6 Functor category

Definition 2.16 (Functor category). Given two categories $\mathcal{C}$ and $\mathcal{D}$, the *functor category* $\mathcal{D}^{\mathcal{C}}$ is the category satisfying the following:

- The objects of $\mathcal{D}^{\mathcal{C}}$ are all functors $\mathcal{C} \to \mathcal{D}$;
- Given two functors $F, G : \mathcal{C} \to \mathcal{D}$, the morphisms from F to G in $\mathcal{D}^{\mathcal{C}}$ are all natural transformations $\theta : F \to G$;
- Composition and identity morphisms in $\mathcal{D}^{\mathcal{C}}$ are defined as follows:
  - The identity morphism $\mathbf{id}_F$ on F is defined as $\theta_X = \mathbf{id}_{F(X)}$ for all $X \in \mathbf{obj}(\mathcal{C})$;
  - The composition of two natural transformations $\theta : F \to G$ and $\phi : G \to H$ is defined as $(\phi \circ \theta)_X = \phi_X \circ \theta_X$ for all $X \in \mathbf{obj}(\mathcal{C})$.

2.2.7 Presheaf category

Definition 2.17 (Presheaf). Given a category $\mathcal{C}$, a *presheaf* on $\mathcal{C}$ is a functor $F : \mathcal{C}^{\mathbf{op}} \to \mathcal{S}et$. A presheaf is a contravariant functor, which means that it reverses the direction of morphisms. In other words, a presheaf is a functor that takes objects in $\mathcal{C}$ and assigns them sets, and takes morphisms in $\mathcal{C}$ and assigns them functions between the corresponding sets. The presheaf F is defined as follows:

- For each $X \in \mathbf{obj}(\mathcal{C})$, $F(X)$ is a set;
- For each $X, Y \in \mathbf{obj}(\mathcal{C})$ and $f \in \mathcal{C}(X, Y)$, $F(f)$ is a function $F(Y) \to F(X)$.

Definition 2.18 (Presheaf category). Given a category $\mathcal{C}$, the *presheaf category* $\widehat{\mathcal{C}}$ is the functor category $\mathcal{S}et^{\mathcal{C}^{\mathbf{op}}}$, which explicitly contains the following:

- The objects of $\widehat{\mathcal{C}}$ are all presheaves on $\mathcal{C}$;
- Given two presheaves $F, G : \mathcal{C}^{\mathbf{op}} \to \mathcal{S}et$, the morphisms from F to G in $\widehat{\mathcal{C}}$ are all natural transformations $\theta : F \to G$.

2.2.8 Yoneda lemma

Definition 2.19 (Yoneda functor). Given a category $\mathcal{C}$, the *Yoneda functor* $よ : \mathcal{C} \to \widehat{\mathcal{C}}$ is defined as follows:

- For each $X \in \mathbf{obj}(\mathcal{C})$, $よ(X)$ is the functor $\mathcal{C}^{\mathbf{op}} \to \mathcal{S}et$ defined as:

$$よ(X)(Y) = \mathcal{C}(Y, X) \tag{2.7}$$

  For all $Y \in \mathbf{obj}(\mathcal{C})$;

- For each $X, Y \in \mathbf{obj}(\mathcal{C})$ and $f \in \mathcal{C}(X, Y)$, $よ(f)$ is the morphism $よ(X) \to よ(Y)$ defined a natural transformation whose component at any given $Z \in \mathcal{C}^{\mathbf{op}}$ is given by:

$$\begin{aligned}(よ(f))_Z : \mathcal{C}(Z, Y) &\to \mathcal{C}(Z, X) \\ g &\mapsto g \circ f\end{aligned} \tag{2.8}$$

  for all $Z \in \mathbf{obj}(\mathcal{C})$.

Theorem 2.20 (Yoneda lemma). For each small[1] category $\mathcal{C}$, the *Yoneda lemma* states that for each object $X \in \mathbf{obj}(\mathcal{C})$ and each presheaf $F \in \hat{\mathcal{C}}$, there exists a natural isomorphism

$$\hat{\mathcal{C}}(よ(X), F) \cong F(X) \tag{2.9}$$

2.2.9 Cartesian closed structure in presheaf categories

Proof Sketch 2.21 (Cartesian closed structure in presheaf categories). Given a small[2] category $\mathcal{C}$, the presheaf category $\hat{\mathcal{C}}$ is a Cartesian closed category.

- The terminal object in $\hat{\mathcal{C}}$ is the constant functor $\mathbf{1} : \mathcal{C}^{\mathbf{op}} \to \mathcal{S}et$, given by

$$\begin{cases}\mathbf{1}(X) = \{*\} & \text{for all } X \in \mathbf{obj}(\mathcal{C}) \\ \mathbf{1}(f) = \mathbf{id}_{\{*\}} & \text{for all } f \in \mathcal{C}(X, Y)\end{cases} \tag{2.10}$$

- The binary product in $\hat{\mathcal{C}}$ is given by the product of functors, which is defined as follows:

$$\begin{aligned}(F \times G)(X) &= F(X) \times G(X) \\ (F \times G)(f) &= F(f) \times G(f)\end{aligned} \tag{2.11}$$

- The exponential in $\hat{\mathcal{C}}$ is derived from the Yoneda lemma,

$$\begin{aligned}G^F(X) &= \hat{\mathcal{C}}(よ(X) \times F, G) \\ G^F(f)(\theta) &= \theta \circ (よ(f) \times \mathbf{id}_F) \quad \text{for all } \theta \in \hat{\mathcal{C}}(よ(Y) \times F, G)\end{aligned} \tag{2.12}$$

2.3 Agda

2.3.1 Basic data types and pattern matching

Here is a simple example selected from PLFA [13] to illustrate the basic data types and pattern matching in Agda. The example is a definition of natural numbers and a plus function.

[1]Here the smallness condition avoids foundational issues analogous to Russell's paradox. While we omit the formal definition of size due to the space limit here, all constructions in this work preserve smallness.

[2]See footnote 1.

```
data ℕ : Set where
  zero : ℕ
  suc : ℕ → ℕ

_+_ : ℕ → ℕ → ℕ
zero + n = n
(suc m) + n = suc (m + n)
```

In Agda we use the data keyword to define a type. Every type has a type, and we use Set to denote the type of all small types. To define a type, we specify its type and its constructors. To define a function, we specify its type and pattern match the input according to the constructors of the type.

2.3.2 Dependent Types

A dependent type is a type that depends on a value. In terms of type judgement, simple types are in form of

$$x_1 : T_1, x_2 : T_2, \dots, x_n : T_n \vdash t(x_1, \dots, x_n) : T$$

In contrast, dependent types are in form of

$$x_1 : T_1, x_2 : T_2, \dots, x_n : T_n \vdash t(x_1, \dots, x_n) : T(x_1, \dots, x_n)$$

or more generally

$$x_1 : T_1, x_2 : T_2(x_1), \dots, x_n : T_n(x_1, \dots, x_{n-1}) \vdash t(x_1, \dots, x_n) : T(x_1, \dots, x_n)$$

A classical example of dependent types in Agda is the type of vectors, which are lists with a length.

```
data Vec (A : Set) : ℕ → Set where
  [] : Vec A zero
  _::_ : ∀{n : ℕ} (x : A) (xs : Vec A n) → Vec A (suc n)
```

Here we define a data type Vec for vectors. The type of vectors is dependent on the length of the vector. With dependent types, we can encode properties directly into types.

2.3.3 Curry-Howard-Lambek correspondence

Curry-Howard correspondence [18] establishes an isomorphism between logic and type theory, where propositions correspond to types and proofs correspond to terms. This correspondence forms the foundation of functional programming languages that can be used to implement proofs. Building upon this, Joachim Lambek showed that cartesian closed categories

provide a natural semantic setting for the simply typed lambda calculus (STLC) [2]. The Curry-Howard-Lambek correspondence can be summarised as follows:

Logic	Type theory	Category theory
Proposition	Type	Object
Proof	Term	Morphism
Falsity	Empty type	Initial object
Truth	Unit type	Terminal object
Implication	Function type	Exponential
Conjunction	Product type	Product
Disjunction	Sum type	Coproduct

Table 2.1: Curry-Howard-Lambek correspondence

2.3.4 Equality, congruence and substitution

Equality here refers to the propositional equality. In Agda it is defined as

```
data _≡_ {A : Set} (x : A) : A → Set where
  refl : x ≡ x
```

With equality as a type, whenever we want to prove that two terms are equal, we need to provide a witness of the equality. For example if we want to prove $x \equiv y$, we write out a term of type x = y, and then give its definition, which constructs a proof of the equality. A simple proof that directly uses the definition of equality is as follows:

```
sym : ∀ {A : Set} {x y : A} → x ≡ y → y ≡ x
sym refl = refl

trans : ∀ {A : Set} {x y z : A} → x ≡ y → y ≡ z → x ≡ z
trans refl refl = refl
```

Here we are able to prove the symmetry and transitivity of equality by using the definition of equality. Those two properties are very useful for later proofs.

We can also have more complex proof with congruence and substitution, which can also be directly derived from the definition of equality as follows:

```
cong : ∀ {A B : Set} (f : A → B) {x y : A} → x ≡ y → f x ≡ f y
cong f refl = refl

subst : ∀ {A : Set} {x y : A} (P : A → Set) → x ≡ y → P x → P y
subst P refl px = px
```

Congruence is a property of equality that states that if two terms are equal, then they can be substituted for each other in any context. Substitution is a property where we can get a new proof by replacing a term in a proof with another term that is equal to it.

Here is a simple example of how congruence can be used in a proof:

```
+-identity : ∀ (n : ℕ) → n + zero ≡ n
+-identity zero = refl
+-identity (suc n) = cong suc (+-identity n)
```

In this example, we want to prove that zero is the right identity of the plus function. We do an inductive proof by pattern matching on the first argument of the plus function.

In the base case, we have zero + zero ≡ zero, which is trivially true by the definition of the plus function (zero is defined to be the left identity of the plus function).

In the inductive case, we need to show suc n + zero ≡ suc n. We can use the definition of the plus function to rewrite the left-hand side as suc (n + zero). By the inductive hypothesis, we know that n + zero ≡ n, so we can substitute n for n + zero in the right-hand side by congruence, and we are done.

2.3.5 Standard library

The Agda standard library [19] is a collection of modules that provide a wide range of useful functions and types. It includes modules for basic data types, such as natural numbers, lists, and vectors. In my implementation, standard library is used as a reference for defining a customised library, which is specified in §3.5.

Builtin pragma

With standard library, we can use actual numbers $1, 2, 3, \ldots$ instead of calling the constructor suc multiple times. This is a convenient feature for testing. Self-defined functions can be used with the builtin numbers with a BUILTIN pragma. This allows me to use a convenient syntax of numbers for self-defined functions.

```
data ℕ : Set where
  zero : ℕ
  suc : ℕ → ℕ
{-# BUILTIN NATURAL ℕ #-}

_+_ : ℕ → ℕ → ℕ
zero + n = n
(suc m) + n = suc (m + n)
{-# BUILTIN NATPLUS _+_ #-}

1+1≡2 : 1 + 1 ≡ 2
1+1≡2 = refl
```

Note that in agda we can use unicode characters for terms, which makes the code more readable. For example, here we simply name the term 1+1≡2.

2.3.6 Interactive programming with holes

A feature of Agda is interactive programming with holes. By leaving holes in place of undefined terms, we can write programs that are incomplete but still type-check, and Agda's type checker will guide completion of the program: the context window displays inferred types of the holes, available variables and candidate terms with their types. Holes also supports case split and refinement, which means we can fill in a hole partially and split it into smaller holes.

In my implementation, I used holes to write the terms in the compiler. The complex terms are incrementally filled and verified by refining partial implementations, reducing post-hoc debugging and ensuring robustness.

2.4 Requirement Analysis

To complete the compiler in Agda we need to implement the following components:

- A file `source.agda` that record the syntax of the source language, which include basic instructions in STLC.
- A file `target.agda` that record the syntax of the target language.
- A file `compiler.agda` that uses `source.agda` and `target.agda` as modules, and write functions whose input is a term in the source language and output is a term in the target language.
- A file `test.agda` that contains test cases for the compiler. The test cases are written in the source language, and the expected output is written in the target language.

The success criteria of the project is that all of the above files are implemented and type checked, which ensures that all the terms are well typed and the compilation is as expected.

Chapter 3

Implementation

Contents

3.1 Tools used **18**
3.2 Repository Overview **18**
3.2.1 Directory structure 18
3.2.2 File dependencies and descriptions 18
3.3 Source Language **19**
3.3.1 Types 19
3.3.2 Contexts, variables and the lookup judgement 20
3.3.3 Terms and the typing judgement 21
3.3.4 Operational Semantics 22
3.4 Target Language **22**
3.4.1 Stack descriptor 23
3.4.2 Grammar 25
3.5 Customised Library **27**
3.5.1 Definition from standard library 27
3.5.2 Subtraction 29
3.5.3 More properties of natural numbers 32
3.6 Compiler **33**
3.6.1 Presheaf semantics 33
3.6.2 Continuations 34
3.6.3 Denotational semantics of types and contexts 35
3.6.4 Functorial mapping 35
3.6.5 Denotational semantics of terms 37
3.6.6 Compilation 41

This chapter details the implementation of the compiler in Agda. Starting with a brief overview of the tools used, it then describes the directory structure and file dependencies. Then it discusses the specification of the source language, and the target language, which lead us to the

problem of implementing a rigorous natural number subtraction and two different approaches. It further reveals the difference between the two approaches and how they affect the implementation of the customised library. Finally a detailed implementation of the compiler is presented.

3.1 Tools used

The project is implemented in Agda 2.7.0.1, which is the latest stable version at the time of writing.

Completing the project is an iterative process. I used Git [20] for version control, and work had been synchronised with a GitHub [21] repository for backup.

For the development environment, I tried both Emacs [22] and Visual Studio Code [23] with an agda-mode extension [24] on Windows Subsystem for Linux with Ubuntu [25] 22.04 LTS. I am more familiar with the snippet and syntax highlighting features of Visual Studio Code, so I used it for most of the development.

Code from the PLFA tutorial and Agda standard library [19] were used as references for implementation of the source language and the customised library. Apart from that, the code is written from scratch.

3.2 Repository Overview

3.2.1 Directory structure

The repository contains two independent implementations of the compiler, organised as follows:

```
.
├── fin
│   ├── lib.agda
│   ├── source.agda
│   ├── target.agda
│   ├── compiler.agda
│   └── test.agda
└── proof
    ├── lib.agda
    ├── source.agda
    ├── target.agda
    ├── compiler.agda
    └── test.agda
```

3.2.2 File dependencies and descriptions

The dependency graph for each implementation is identical as follows:



Figure 3.1: Dependency graph for the compiler

File	Description
`lib.agda`	shared utilities (e.g. natural numbers and operations, equality, etc.)
`source.agda`	source language syntax
`target.agda`	target language (stack machine) instructions
`compiler.agda`	compiler derived from denotational semantics
`test.agda`	test cases for the compiler

Table 3.1: File descriptions

3.3 Source Language

A simply typed lambda calculus (STLC) is a typed lambda calculus with only one type constructor ($\Rightarrow$) [1] that builds function types.

3.3.1 Types

In this dissertation, the source language is an STLC with the following primitive types:

- comm: the commands
- intexp: the integer expressions
- intacc: the integer acceptors
- intvar: the integer variables

and the set Θ of types is defined as follows:

$$\Theta := \mathsf{comm} \mid \mathsf{intexp} \mid \mathsf{intacc} \mid \mathsf{intvar} \mid \Theta \Rightarrow \Theta$$

[1] Normally we use $\rightarrow$ to denote the function type, but here we use $\Rightarrow$ to be consistent with the notation in Agda, as $\rightarrow$ is a primitive in Agda.

which corresponds with the following agda implementation:

```
data Type : Set where
  comm : Type
  intexp : Type
  intacc : Type
  intvar : Type
  _⇒_ : Type → Type → Type
```

Subtypes

We use the preorder $A \leq: B$ to denote the subtype relation A is a subtype of B, as it has the following properties:

- reflexivity: $A \leq: A$;
- transitivity: if $A \leq: B$ and $B \leq: C$, then $A \leq: C$.

The source language has the following subtype relations:

$$\text{intvar} \leq: \text{intexp} \qquad \text{intvar} \leq: \text{intacc}$$

since a variable can be used as an expression (e.g. $\mathsf{x} + 1$) or an acceptor (e.g. $\mathsf{x} := 1$).

For function types we have the contravariant subtyping:

$$A' \leq: A \wedge B \leq: B' \Rightarrow A \to B \leq: A' \to B'$$

it is defined in Agda as follows:

```
data _≤:_ : Type → Type → Set where
  ≤:-refl : ∀{A} → A ≤: A
  ≤:-trans : ∀{A A' A"} → A ≤: A' → A' ≤: A" → A ≤: A"
  ≤:-fn : ∀{A A' B B'} → A' ≤: A → B ≤: B' → A ⇒ B ≤: A' ⇒ B'

  var-≤:-exp : intvar ≤: intexp
  var-≤:-acc : intvar ≤: intacc
```

3.3.2 Contexts, variables and the lookup judgement

We define the context as a finite list of types. When we are looking up a variable, the type of the variable is either the head of the context or in the tail of the context. The definition in

Agda is as follows:

```
-- Contexts
data Context : Set where
  · : Context
  _,_ : Context → Type → Context

-- Variables and the lookup judgement
data _∈_ : Type → Context → Set where
  Zero : ∀{Γ A} → A ∈ Γ , A
  Suc : ∀{Γ A B} → B ∈ Γ → B ∈ Γ , A
```

where the Zero case corresponds to the situation where the variable is the head of the context, and the Suc case corresponds to the situation where the variable is in the tail of the context.

3.3.3 Terms and the typing judgement

The terms and the typing judgement are described in the following rules:

- For any variable we have

$$\frac{a : A \in \Gamma}{\Gamma \vdash a : A}\ \textsc{Var}$$

Figure 3.2: Variable typing rule

- For subtyping we have

$$\frac{\Gamma \vdash a : A \qquad A \leq: B}{\Gamma \vdash a : B}\ \textsc{Sub}$$

Figure 3.3: Subtyping rule

- For lambda abstraction we have

$$\frac{\Gamma, x : A \vdash e : B}{\Gamma \vdash \lambda x : A.e : A \Rightarrow B}\ \textsc{Lambda} \qquad \frac{\Gamma \vdash e : A \Rightarrow B \qquad \Gamma \vdash e' : A}{\Gamma \vdash e\, e' : B}\ \textsc{App}$$

Figure 3.4: Lambda abstraction typing rules

- For command we have

$$\frac{}{\Gamma \vdash \texttt{skip} : \text{comm}}\ \text{Skip} \qquad \frac{\Gamma \vdash c_1 : \text{comm} \qquad \Gamma \vdash c_2 : \text{comm}}{\Gamma \vdash c_1 ; c_2 : \text{comm}}\ \textsc{Seq} \qquad \frac{\Gamma, x : \text{intvar} \vdash c : \text{comm}}{\Gamma \vdash \texttt{new}\ x\ \texttt{in}\ c : \text{comm}}\ \textsc{NewVar}$$

Figure 3.5: Command typing rules

- For integer expression we have

$$\frac{z : \mathbb{Z}}{\Gamma \vdash \texttt{lit}\ z : \text{intexp}}\ \textsc{Lit} \qquad \frac{\Gamma \vdash e : \text{intexp}}{\Gamma \vdash -e : \text{intexp}}\ \textsc{Neg} \qquad \frac{\Gamma \vdash e_1 : \text{intexp} \quad \Gamma \vdash e_2 : \text{intexp}}{\Gamma \vdash e_1 + e_2 : \text{intexp}}\ \textsc{Plus}$$

Figure 3.6: Integer expression typing rules

The corresponding Agda implementation is as follows, note that the Agda implementation only does not include the name of the terms, as a term can be specified by how it is constructed (i.e. the name of the typing rules and the corresponding types)

```
data _⊢_ : Context → Type → Set where
  Var : ∀{Γ A} → A ∈ Γ → Γ ⊢ A

  -- subtyping
  Sub : ∀{Γ A B} → Γ ⊢ A → A <: B → Γ ⊢ B

  -- lambda function and application
  Lambda : ∀{Γ A B} → Γ , A ⊢ B → Γ ⊢ A ⇒ B
  App : ∀{Γ A B} → Γ ⊢ A ⇒ B → Γ ⊢ A → Γ ⊢ B

  -- command
  Skip : ∀{Γ} → Γ ⊢ comm
  Seq : ∀{Γ} → Γ ⊢ comm → Γ ⊢ comm → Γ ⊢ comm
  NewVar : ∀{Γ} → Γ , intvar ⊢ comm → Γ ⊢ comm
  Assign : ∀{Γ} → Γ ⊢ intacc → Γ ⊢ intexp → Γ ⊢ comm

  -- intexp
  Lit : ∀{Γ} → ℤ → Γ ⊢ intexp
  Neg : ∀{Γ} → Γ ⊢ intexp → Γ ⊢ intexp
  Plus : ∀{Γ} → Γ ⊢ intexp → Γ ⊢ intexp → Γ ⊢ intexp
```

3.3.4 Operational Semantics

The operational semantics of the source language is defined with renaming, substitution and reduction rules. However, since the compiler itself does not require the operational semantics, the implementation is included in Appendix A. Operational semantics can be used to verify the correctness of the compiler in the future, but it is beyond the scope of this project.

3.4 Target Language

The target language is an assembly-style intermediate language for a stack machine. It is defined with four stack-descriptor-indexed families of non-terminals:

- $<\mathsf{L}_{sd}>$: left-hand sides
- $<\mathsf{S}_{sd}>$: simple right-hand sides

- $<\mathsf{R}_{sd}>$: right-hand sides
- $<\mathsf{I}_{sd}>$: instruction sequences

The intent of the indexing here is any instructions with the form $<\mathsf{I}_{sd}>$ is an instruction sequence that can be executed when the current stack descriptor is sd.

3.4.1 Stack descriptor

The stack descriptor sd is defined as a pair of natural numbers $\langle f, d \rangle$, where f is the frame number and d is the displacement.

It is defined in Agda as follows:

```
record SD : Set where
  constructor ⟨_,_⟩
  field
    f : ℕ
    d : ℕ
```

For a stack descriptor sd, we can use SD.f sd to access the frame number and SD.d sd to access the displacement.

Order

The stack descriptor is ordered lexicographically with $\leq_\mathsf{s}$ as follows:

$$\langle f, d \rangle \leq_\mathsf{s} \langle f', d' \rangle \Leftrightarrow f < f' \vee (f = f' \wedge d \leq d')$$

It is defined in Agda as follows:

```
data _≤ₛ_ : SD → SD → Set where
  <-f : ∀ {f f' d d'} → f < f' → ⟨ f , d ⟩ ≤ₛ ⟨ f' , d' ⟩
  ≤-d : ∀ {f d d'} → d ≤ d' → ⟨ f , d ⟩ ≤ₛ ⟨ f , d' ⟩
```

Here definition of the order of natural numbers is in the customised library. Please refer to §3.5 for more details.

Addition and subtraction of stack descriptors

We define addition and subtraction of stack descriptors as follows:

$$\langle f, d \rangle \pm_s n = \langle f, d \pm n \rangle \text{ for } n \in \mathbb{N}$$

Addition of stack displacement is defined in Agda as follows:

```
_+ₛ_ : SD → ℕ → SD
⟨ f , d ⟩ +ₛ n = ⟨ f , d + n ⟩
```

For subtraction, we need to make sure that the subtraction is valid, i.e. the subtracted displacement adjustment is less than or equal to the displacement of the current stack descriptor. This leads us to the problem of defining the subtraction of natural numbers, which is explained in detail in §3.5.2

Properties of order and addition

We can use reflexivity and transitivity of the order $<$ and $\leq$ to prove that the order of stack descriptors $\leq_s$ is a preorder. The proof is as follows:

Proof. We need to show that $\leq_s$ is reflexive and transitive.

- Reflexivity: $sd \leq_s sd$ is trivially true since $f = f$ and $d \leq d$.
- Transitivity: if $\langle f, d\rangle \leq_s \langle f', d'\rangle$ and $\langle f', d'\rangle \leq_s \langle f'', d''\rangle$, we do the following case split:
  - If $f < f'$, $f' = f''$ and $d \leq d'$, then $f < f''$;
  - If $f < f'$ and $f' < f''$, then $f < f''$ by transitivity of $<$;
  - If $f = f'$, $d \leq d'$ and $f' < f''$, then $f < f''$;
  - If $f = f'$, $d \leq d'$, $f' = f''$ and $d' \leq d''$, then $f = f''$ and $d \leq d''$ by transitivity of $\leq$. [2]

□

The proof is defined in Agda as follows:

```
≤ₛ-refl : ∀{sd : SD} → sd ≤ₛ sd
≤ₛ-refl {⟨ f , d ⟩} = ≤-d ≤-refl

≤ₛ-trans : ∀{sd sd' sd'' : SD} → sd ≤ₛ sd' → sd' ≤ₛ sd'' → sd ≤ₛ sd''
≤ₛ-trans (<-f f<f') (≤-d _) = <-f f<f'
≤ₛ-trans (<-f f<f') (<-f f'<f'') = <-f (<-trans f<f' f'<f'')
≤ₛ-trans (≤-d _) (<-f f'<f'') = <-f f'<f''
≤ₛ-trans (≤-d d≤d') (≤-d d'≤d'') = ≤-d (≤-trans d≤d' d'≤d'')
```

When a natural number is added to a stack descriptor, it is guaranteed that the new stack descriptor is not less than the old stack descriptor. This can be proved directly using the same properties of integer addition. The proof is defined in Agda as follows:

```
+ₛ→≤ₛ : ∀{sd : SD} → ∀{n : ℕ} → sd ≤ₛ sd +ₛ n
+ₛ→≤ₛ = ≤-d +→≤
```

Proofs here are based on the properties of natural numbers specified in the customised library. Please refer to §3.5 for more details.

[2] Note that the properties of $=$ is not required in as in the Agda definition of $\leq_s$, we directly define f' to be f instead of defining a equivalence relation, thus equivalence can be checked directly via type checking.

3.4.2 Grammar

The grammar of the target language is defined as follows, given the current stack descriptor $sd = \langle f, d \rangle$, and new stack descriptor sd' and sd^v:

$$
\begin{aligned}
&\langle \mathsf{L}_{sd} \rangle ::= sd^v \quad \text{when } sd^v \leq_{\mathrm{s}} sd \\
&\qquad\quad \mid \texttt{sbrs} \\
&\langle \mathsf{S}_{sd} \rangle ::= \langle \mathsf{L}_{sd} \rangle \\
&\qquad\quad \mid \texttt{lit } \langle \text{integer} \rangle \\
&\langle \mathsf{R}_{sd} \rangle ::= \langle \mathsf{S}_{sd} \rangle \\
&\qquad\quad \mid \langle \text{unary operator} \rangle \langle \mathsf{S}_{sd} \rangle \\
&\qquad\quad \mid \langle \mathsf{S}_{sd} \rangle \langle \text{binary operator} \rangle \langle \mathsf{S}_{sd} \rangle \\
&\langle \mathsf{I}_{sd} \rangle ::= \texttt{stop} \\
&\qquad\quad \mid \langle \mathsf{L}_{sd+\delta} \rangle := \langle \mathsf{R}_{sd} \rangle [\delta] \,;\, \langle \mathsf{I}_{sd+\delta} \rangle \\
&\qquad\quad \mid \texttt{if } \langle \mathsf{S}_{sd} \rangle \langle \text{relational operator} \rangle \langle \mathsf{S}_{sd} \rangle [\delta] \\
&\qquad\qquad \texttt{then } \langle \mathsf{I}_{sd+\delta} \rangle \texttt{ else } \langle \mathsf{I}_{sd+\delta} \rangle \\
&\qquad\quad \mid \texttt{adjustdisp}[\delta] \,;\, \langle \mathsf{I}_{sd+\delta} \rangle \\
&\qquad\quad \mid \texttt{popto } sd' \,;\, \langle \mathsf{I}_{sd'} \rangle \quad \text{when } sd' \leq_{\mathrm{s}} sd
\end{aligned}
\Bigg\} \text{if } d + \delta \geq 0
$$

Figure 3.7: Target language grammar

Some explanations of the grammar are as follows:

- `sbrs`: a register used to communicate the result of function procedures implemented by closed subroutines.
- `lit`: a literal integer.
- `stop`: stop the execution of the program.
- δ: a displacement adjustments, which can be any integer that satisfies the condition mentioned above.
- `adjustdisp`: adjust the displacement of specified stack descriptor.
- `popto`: pop to the specified stack descriptor, which is used to reduce the number of frames.

Note that the design of right-hand sides ensures that the right operand of the assignment contains at most one operator, which means that any instruction that involves multiple operators must be divided into multiple instructions. This feature of the target language significantly complicates the compilation of integer expressions, which we will analysis in detail in §3.6.5.

Operators

The operators are defined as follows:

$$\begin{aligned} \text{unary operator} &\in \{\texttt{UNeg}\} \\ \text{binary operator} &\in \{\texttt{BPlus}, \texttt{BMinus}, \texttt{BTimes}\} \\ \text{relational operator} &\in \{\texttt{RLeq}, \texttt{RLt}\} \end{aligned}$$

The operators are defined in Agda as follows:

```
data UnaryOp : Set where
  UNeg : UnaryOp

data BinaryOp : Set where
  BPlus : BinaryOp
  BMinus : BinaryOp
  BTimes : BinaryOp

data RelOp : Set where
  RLeq : RelOp
  RLt : RelOp
```

Left-hand sides, simple right-hand sides and right-hand sides

The left-hand sides, simple right-hand sides and right-hand sides are straightforward, and they are defined as follows in Agda:

```
-- Lefthand sides
data L (sd : SD) : Set where
  l-var : (sd' : SD) → sd' ≤ₛ sd → L sd
  l-sbrs : L sd

-- Simple righthand sides
data S (sd : SD) : Set where
  s-l : L sd → S sd
  s-lit : ℤ → S sd

-- Righthand sides
data R (sd : SD) : Set where
  r-s : S sd → R sd
  r-unary : UnaryOp → S sd → R sd
  r-binary : S sd → BinaryOp → S sd → R sd
```

Instruction sequences

In the instruction sequences, δ as a displacement adjustment can be positive, zero or negative. To make a rigorous definition, we define δ as a natural number, and treat the positive and

negative displacements as two different instructions, using addition and subtraction of stack descriptors respectively. Since the definition of negative displacement involves the rigorous definition of subtraction, we will discuss the subtraction of natural numbers and corresponding implementation in §3.5.2. The definition (except for the negative displacement) in Agda is as follows:

```
data I (sd : SD) : Set where
  stop : I sd
  assign-inc : (δ : ℕ) → L (sd +ₛ δ) → R sd → I (sd +ₛ δ) → I sd
  if-then-else-inc : (δ : ℕ) → S sd → RelOp → S sd
                     → I (sd +ₛ δ) → I (sd +ₛ δ) → I sd
  adjustdisp-inc : (δ : ℕ) → I (sd +ₛ δ) → I sd
  popto : (sd' : SD) → sd' ≤ₛ sd → I sd' → I sd
```

3.5 Customised Library

The customised library is a collection of types and functions centred around natural numbers, their operations and properties. Instead of using the standard library [19], I implemented a customised library for the following reasons:

- **Specialised Requirements**: The standard library's definitions (e.g. subtraction as monus, '$\dot{-}$') do not align with my project's need for a rigorously defined subtraction operation. This is specified in §3.5.2.
- **Verification Clarity**: By defining the basic data types and functions from scratch, I can tailor properties (e.g. the property '$n - [n - m] \equiv m$') to my specific requirements, and avoid unnecessary complexity (e.g. '$\leq$' being also defined as a boolean predicate in the standard library).
- **Minimal Dependencies**: Avoiding the standard library reduces external assumptions, making the project self-contained.

3.5.1 Definition from standard library

The following is a list of definitions (or their variants) from the standard library and PLFA [13] that are used in the customised library:

- Natural numbers ($\mathbb{N}$), integers ($\mathbb{Z}$) and addition ($+$).
- Equality ($\equiv$), congruence (cong), substitution (sub) and transitivity of equality (trans).
- Order of natural numbers, including $\leq$ and $<$.
- Some properties of order and addition.

The properties of order and addition contains the following terms summarised in the table below:

Agda term	Mathematical meaning
+-identity^r	$\forall n \in \mathbb{N}.\ n + 0 = n$
+-suc^r	$\forall m, n \in \mathbb{N}.\ m + \mathsf{suc}\ n = \mathsf{suc}\ (m + n)$
+-comm	$\forall m, n \in \mathbb{N}.\ m + n \equiv n + m$
≤-irrelevant	For all $m, n \in \mathbb{N}$, we consider all proofs of $m \leq n$ to be equal.
≤-refl	$\forall n \in \mathbb{N}.\ n \leq n$
≤-trans	$\forall m, n, p \in \mathbb{N}.\ m \leq n \wedge n \leq p \Rightarrow m \leq p$
n≤suc-n	$\forall n \in \mathbb{N}.\ n \leq \mathsf{suc}\ n$
m≡n, p≤n→p≤m	$\forall p, m, n \in \mathbb{N}.\ m \equiv n \wedge p \leq n \Rightarrow p \leq m$
+→≤	$\forall m, n \in \mathbb{N}.\ m \leq m + n$
+→≤^{r}	$\forall m, n \in \mathbb{N}.\ m \leq n + m$
<-trans	$\forall m, n, p \in \mathbb{N}.\ m < n \wedge n < p \Rightarrow m < p$
<→s≤	$\forall m, n \in \mathbb{N}.\ m < n \Rightarrow \mathsf{suc}\ m \leq n$
<→≤	$\forall m, n \in \mathbb{N}.\ m < n \Rightarrow m \leq n$

Table 3.2: Properties of order and addition

The detailed definitions are included in Appendix [TBD].

Totality of order

It is useful to case split the order of natural numbers, i.e. for any $m, n \in \mathbb{N}$, we have $m \leq n$ or $n \leq m$. This is defined in Agda by defining either case as an instance of the Total type, and then showing that for any $m, n \in \mathbb{N}$, we can construct a Total type by case-split and with-construct. The proof is as follows:

```
data Order : ℕ → ℕ → Set where
  leq : ∀ {m n : ℕ} → m ≤ n → Order m n
  geq : ∀ {m n : ℕ} → n ≤ m → Order m n

≤-compare : ∀ {m n : ℕ} → Order m n
≤-compare {zero} {n} = leq z≤n
≤-compare {suc m} {zero} = geq z≤n
≤-compare {suc m} {suc n} with ≤-compare {m} {n}
```

3.5.2 Subtraction

In the standard library [19], the subtraction of natural numbers is defined as follows:[3]

```
_∸_ : ℕ → ℕ → ℕ
n ∸ zero = n
zero ∸ suc m = zero
suc n ∸ suc m = n ∸ m
{-# BUILTIN NATMINUS _∸_ #-}

0∸1≡0 : 0 ∸ 1 ≡ 0
0∸1≡0 = refl
```

Mathematically we have $n \dot{-} m = \max(0, n - m)$. This definition is valid and ensures that the result is a natural number. However, there are two problems with this definition for our implementation:

- The definition of subtraction is not rigorous. Input n should be guaranteed not to be less than m, and at any point in the program if there is a subtraction where $n < m$, the program should fail to type check instead of making 0 type check with it.
- We lose many numerical properties of natural numbers. For example, $n \dot{-} m + m = n$ or $n \dot{-} (n \dot{-} m) = m$ are not guaranteed to hold if we allow $n < m$. Those properties are very useful for the implementation of the compiler.

We need a more rigorous definition of subtraction that ensures the result is a natural number and preserves the numerical properties. The key is to have a proof that the first argument is greater than or equal to the second argument. There are two approaches to achieve this:

Explicit proof-passing approach

A straightforward approach is directly include a third argument in the subtraction function, which is a proof that the first argument is greater than or equal to the second argument. The definition in Agda is as follows:

```
_−_ : (n : ℕ) → (m : ℕ) → (p : m ≤ n) → ℕ
(n − zero) (z≤n) = n
(suc n − suc m) (s≤s m≤n) = (n − m) m≤n
```

The corresponding implementation for subtraction of stack descriptors carries along the proof as follows:

```
_−ₛ_ : (sd : SD) → (n : ℕ) → (n≤d : n ≤ SD.d sd) → SD
(⟨ f , d ⟩ −ₛ n) n≤d = ⟨ f , (d − n) n≤d ⟩
```

[3] The definition in the standard library directly uses symbol $-$, but here for clarity we use $\dot{-}$ instead.

In the implementation of instruction sequences, we need to carry along the proof as well. The implementation in Agda is as follows:

```
data I (sd : SD) : Set where
  assign-dec : (δ : ℕ) → (δ≤d : δ ≤ SD.d sd) → L ((sd −ₛ δ) δ≤d)
                 → R sd → I ((sd −ₛ δ) δ≤d) → I sd
  if-then-else-dec : (δ : ℕ) → (δ≤d : δ ≤ SD.d sd)
                       → S sd → RelOp → S sd
                       → I ((sd −ₛ δ) δ≤d)
                       → I ((sd −ₛ δ) δ≤d) → I sd
  adjustdisp-dec : (δ : ℕ) → (δ≤d : δ ≤ SD.d sd)
                     → I ((sd −ₛ δ) δ≤d) → I sd
```

Fin-based approach

Instead of include the proof directly, we can define the type of the subtrahend to be dependent on the minuend, where the dependence encodes the proof. There is a standard library in Agda on finite sets, where a type Fin is defined as a type that depends on a natural number n, where Fin n can be seen as the set of natural numbers less than n. The definition is as follows:

```
data Fin : ℕ → Set where
  fzero : ∀ {n} → Fin (suc n)
  fsuc : ∀ {n} → Fin n → Fin (suc n)
```

An intuitive explanation of the definition of Fin is as follows:

- Base case: when $n = 0$, we do not have any constructor that gives us an element of type Fin 0, corresponding to the empty set.
- Inductive case:



Figure 3.8: Inductive case of Fin

For subtraction, we only need to change the type of the subtrahend to Fin (suc n) compared to the implementation of addition. The definition in Agda is as follows:

```
_−_ : (m : ℕ) → Fin (suc m) → ℕ
m − fzero = m
suc m − fsuc n = m − n
```

The corresponding implementation for subtraction of stack descriptors and instruction sequences only requires a change of the type of the subtrahend as follows:

```
_−ₛ_ : (sd : SD) → Fin (suc (SD.d sd)) → SD
⟨ f , d ⟩ −ₛ n = ⟨ f , d − n ⟩

-- Instruction sequences
data I (sd : SD) : Set where
assign-dec : (δ : Fin (suc (SD.d sd))) → L (sd −ₛ δ) → R sd
                → I (sd −ₛ δ) → I sd
if-then-else-dec : (δ : Fin (suc (SD.d sd))) → S sd → RelOp → S sd
                      → I (sd −ₛ δ) → I (sd −ₛ δ) → I sd
adjustdisp-dec : (δ : Fin (suc (SD.d sd))) → I (sd −ₛ δ) → I sd
```

Comparison of the two approaches

The explicit approach carries proofs directly in the type system, which is more transparent but verbose. The Fin-based approach initially appears to be more elegant, as it encapsulates bounds check within a refined type. However, this elegance is superficial: constructing a term of type Fin n still requires the same underlying proof of boundedness, shifting complexity to auxiliary conversions.

To prove a property with the Fin-based approach, we need the following auxiliary function:

```
≤→Fin : ∀ {m n} → m ≤ n → Fin (suc n)
≤→Fin z≤n = fzero
≤→Fin (s≤s p) = fsuc (≤→Fin p)
```

Here is a comparison of the two approaches representing the property suc $(n-m) \equiv$ suc $n-m$:

```
-- Explicit approach
−-suc : ∀ {n m} → {m≤n : m ≤ n}
          → suc ((n − m) m≤n) ≡ (suc n − m) (≤-trans m≤n n≤suc-n)

-- Fin-based approach
−-suc : ∀{n m} → {m≤n : m ≤ n}
      → suc (n − ≤→Fin m≤n) ≡ suc n − ≤→Fin (≤-trans m≤n n≤suc-n)
```

Figure 3.9: Comparison of explicit vs. Fin-based approaches

The explicit approach's verbosity pays off in readability when representing properties. the type of the term directly mirrors the mathematical statement with an explicit proof. In contrast, the Fin-based approach fractures this relationship, and reconstructing the original subtrahend requires tracing the type of the proof. This becomes cumbersome when the proofs are chained inequalities with transitivity, making the code harder to understand.

Although the project completed both approaches, the following implementation of the compiler uses the explicit proof-passing approach due to its clarity.

3.5.3 More properties of natural numbers

Given the definition of subtraction, the additional properties of natural numbers are summarised in the table below:

Agda term	Mathematical meaning
—-irrelevant	For all $m, n \in \mathbb{N}$, we consider all terms of $n - m$ with different proofs of $m \leq n$ to be equal.
—→≤	$\forall m, n \in \mathbb{N}.\ m \leq n \Rightarrow m - n \leq m$
n—n≡0	$\forall n \in \mathbb{N}.\ n - n \equiv 0$
—-suc	$\forall m, n \in \mathbb{N}.\ \mathsf{suc}\ (n - m) \equiv \mathsf{suc}\ n - m$
n—[n—m]≡m	$\forall m, n \in \mathbb{N}.\ m \leq n \Rightarrow n - (n - m) \equiv m$
n+m—m≡n	$\forall m, n \in \mathbb{N}.\ n + m - m \equiv n$
suc-d≤d′→d′—[d′—[suc-d]]	$\forall d, d' \in \mathbb{N}.\ \mathsf{suc}\ d \leq d' \Rightarrow d \leq d' - (d' - \mathsf{suc}\ d)$

Table 3.3: Properties of natural numbers related to subtraction

The equations in the table above are used later in the implementation of the compiler to prove that for instructions with displacement, the given displacement is valid. Here is the mathematical proof of the property n—[n—m]≡m:

Proof. We need to show that $n - (n - m) \equiv m$ for all $m, n \in \mathbb{N}$ such that $m \leq n$. We do the following case split:

- If $m = 0$, then the term reduces to $n - n \equiv 0$, which is a term we have already proved.
- Otherwise it is guaranteed that $m = \mathsf{suc}\ m'$ for some $m' \in \mathbb{N}$, and $n = \mathsf{suc}\ n'$ for some $n' \in \mathbb{N}$, because $m \leq n$. The term reduces to $\mathsf{suc}\ n' - (n' - m') \equiv \mathsf{suc}\ m'$. With congruence, we can reduce the term to $n' - (n' - m') \equiv m'$, which is the inductive hypothesis.

□

The full implementation of the properties is included in Appendix [TBD].

3.6 Compiler

3.6.1 Presheaf semantics

The denotation semantics interprets types as presheaves over a base category Σ.

Base category

The base category Σ is specified as follows:

- Objects are stack descriptors ($sd = \langle f, d \rangle$);
- Morphisms are stack expansions ($\iota : sd \to sd'$ for $sd \leq_{\mathsf{s}} sd'$).

We define SD as the set of stack descriptors. Then the base category is the category determined by preorder $\underline{\mathsf{SD}} = \{\mathsf{SD}, \leq_{\mathsf{s}}\}$ in Example 2.4.

Semantic category

Semantic category $\mathcal{K}$ is the functor category $\mathrm{PDOM}^{\Sigma^{\mathbf{op}}}$, where PDOM is the category of predomains and continuous functions.

- Objects are presheaves: $P = \Sigma^{\mathbf{op}} \to \mathrm{PDOM}$;
- Morphisms are natural transformations: $\eta : P \to Q$ for $P, Q \in \mathrm{PDOM}^{\Sigma^{\mathbf{op}}}$.

The proof for semantic category is similar to the one in §2.2.9, and the full proof is given by Oles in [4, 5].

- Products are given by pointwise product of presheaves: $P \times Q = \lambda sd.\ P\ sd \times Q\ sd$.
- Exponentials $P \Rightarrow_s Q$ are given by the following equation:

$$
\begin{aligned}
P \Rightarrow_s Q &= \lambda sd.\ \mathcal{K}(\text{ﻉ}(sd) \times P, Q) \\
&\cong \lambda sd.\{F \mid sd' \in \Sigma^{\mathbf{op}}, F : (\text{ﻉ}(sd) \times P)\ sd' \to Q\ sd'\} \\
&\qquad\qquad \text{by definition of morphisms in Definition 2.1} \\
&\cong \lambda sd.\{F \mid sd' \in \Sigma^{\mathbf{op}}, F : (\Sigma^{\mathbf{op}}(sd', sd) \times P)\ sd' \to Q\ sd'\} \\
&\qquad\qquad \text{by definition of ﻉ in Definition 2.19} \qquad (3.1) \\
&\cong \lambda sd.\{F \mid sd' \in \Sigma, F : (\Sigma(sd, sd') \times P)\ sd' \to Q\ sd'\} \\
&\qquad\qquad \text{by definition of opposite category in Definition 2.5} \\
&\cong \lambda sd.\{F \mid sd' \in \Sigma, F : sd \leq_{\mathsf{s}} sd' \to P\ sd' \to Q\ sd'\} \\
&\qquad\qquad \text{by definition of } \Sigma \text{ above}
\end{aligned}
$$

The definition of products and exponentials in Agda is as follows:

```
_×ₛ_ : (SD → Set) → (SD → Set) → SD → Set
(P ×ₛ Q) sd = P sd × Q sd

_⇒ₛ_ : (SD → Set) → (SD → Set) → SD → Set
(P ⇒ₛ Q) sd = ∀{sd'} → (sd ≤ₛ sd') → P sd' → Q sd'
```

The definition of exponentials captures the essence of stack-respecting computations in the compiler.

- $sd \leq_{\mathsf{s}} sd'$ means the new stack descriptor sd' is larger than the old stack descriptor sd, representing the dynamic expansion of the stack during execution.
- P and Q are denotational semantics of types.

The exponential $P \Rightarrow_{\mathsf{s}} Q$ is a function that given a proof of stack growth, transforming a P-value at sd' into a Q-value at sd', and the $\forall sd'$ quantifier guarantees it works no matter how the stack grows. It is designed in this way so that a functor $[\![-]\!] : \Theta \to \mathcal{K}$ exists that interprets the type constructor $\Rightarrow$ as the exponential functor $\Rightarrow_s$.

3.6.2 Continuations

In Reynolds' model, we construct Compl and Intcompl[4] as tools in the denotational semantics to represent command continuations and integer continuations, respectively. We can think of integer continuations as a function that takes an integer and returns a command continuation. Mathematically we have

$$\mathsf{Compl}\ sd : \mathsf{SD} \to O \qquad \mathsf{Intcompl}\ sd : \mathbb{Z} \to (\mathsf{SD} \to O)$$

where O is an unspecified domain of outputs.

For code generation we define the continuation to be the instruction sequence in the target language:

$$\mathsf{Compl}\ sd = \mathsf{I}_{sd}$$

The definition of integer continuation is an exponential object in the presheaf category, which is defined as follows:

$$\mathsf{Intcompl} = R \Rightarrow_s \mathsf{Compl}$$

where R is the functor corresponding to the right-hand sides in the target language, $R\ sd = \mathsf{R}_{sd}$. This works because the right-hand sides are all integer expressions we can use according to the grammar of the target language.

Thus we can directly use the definition of R in the target language to define the integer continuation as follows:

```
Compl : SD → Set
Compl sd = I sd

Intcompl : SD → Set
Intcompl = R ⇒ₛ Compl
```

[4] In the work of Reynolds, compl and intcompl are introduced as new types in the source language, and the definitions here are the denotational semantics of those types, respectively. Since these two introduced types are only used for the denotational semantics, I think directly defining them as denotational semantics constructs is natural and sufficient.

3.6.3 Denotational semantics of types and contexts

With the idea of continuation, the denotational semantics of types can be defined as follows:

- Command passes on command continuations;
- Integer expressions takes an integer continuation and returns a command continuation;
- Integer acceptors take a command continuation and returns an integer continuation;
- Integer variables can be used both as an integer expression and an integer acceptor, so it has the denotational semantics of both, defined as a product of the two.

The denotational semantics of types in Agda as follows:

```
⟦_⟧ty : Type → SD → Set
⟦ comm ⟧ty = Compl ⇒ₛ Compl
⟦ intexp ⟧ty = Intcompl ⇒ₛ Compl
⟦ intacc ⟧ty = Compl ⇒ₛ Intcompl
⟦ intvar ⟧ty = ⟦ intexp ⟧ty ×ₛ ⟦ intacc ⟧ty
⟦ θ₁ ⇒ θ₂ ⟧ty = ⟦ θ₁ ⟧ty ⇒ₛ ⟦ θ₂ ⟧ty
```

The denotational semantics of contexts is essentially a list expressed by the product of the denotational semantics of the types in the context. The denotational semantics of contexts is defined as follows, where we use $\emptyset$ to denote the denotational semantics of the empty context:

```
-- Unit type for empty context
data ∅ : Set where
  unit : ∅

-- Context Interpretation
⟦_⟧ctx : Context → SD → Set
⟦ · ⟧ctx _ = ∅
⟦ Γ , A ⟧ctx sd = ⟦ Γ ⟧ctx sd × ⟦ A ⟧ty sd
```

3.6.4 Functorial mapping

The denotational semantics of terms is defined as a functor, defined in §2.14. Before we define its object mapping, we need to define the functorial mapping, which acts on the morphisms of the base category Σ.

Since the morphisms of the base category Σ are stack expansions, the functorial mapping essentially lifts a semantic object indexed by a stack descriptor sd to a semantic object indexed by a larger stack descriptor sd'. This ensures the denotational semantics of terms remains valid when the stack grows.

Functorial mapping for exponentials, products, contexts and types

Functorial mapping for exponentials is simple, as the original exponential can transforms a P-value at any sd' greater than sd'' into a Q-value at sd'. To construct a new exponential, we simply take a new P-value at ad'' which is greater than sd', and use the original exponential to transform it into a Q-value at sd', as sd'' is guaranteed to be greater than sd due to the transitivity of order of the stack descriptors.

Functorial mapping for products is defined pointwise.

Thus, functorial mapping of types can be defined simply use the definition of functorial mapping of products and exponentials, as denotational semantics of types is defined as either product or exponential. As a list of types, the functorial mapping of contexts is simply a map of the functorial mapping of types in the list. The definition of functorial mapping in Agda is as follows:

```
fmap-⇒ : ∀ {P Q sd sd'} → (P ⇒ₛ Q) sd → sd ≤ₛ sd' → (P ⇒ₛ Q) sd'
fmap-⇒ P⇒Q sd≤ₛsd' sd'≤ₛsd'' p = P⇒Q (≤ₛ-trans sd≤ₛsd' sd'≤ₛsd'') p

fmap-ty : ∀ {A sd sd'} → ⟦ A ⟧ty sd → sd ≤ₛ sd' → ⟦ A ⟧ty sd'
fmap-ty {comm} = fmap-⇒ {Compl} {Compl}
fmap-ty {intexp} = fmap-⇒ {Intcompl} {Compl}
fmap-ty {intacc} = fmap-⇒ {Compl} {Intcompl}
fmap-ty {intvar} ( exp , acc ) sd≤ₛsd' =
  ( fmap-ty {intexp} exp sd≤ₛsd' , fmap-ty {intacc} acc sd≤ₛsd' )
fmap-ty {A ⇒ B} = fmap-⇒ {⟦ A ⟧ty} {⟦ B ⟧ty}

fmap-ctx : ∀ {Γ sd sd'} → ⟦ Γ ⟧ctx sd → sd ≤ₛ sd' → ⟦ Γ ⟧ctx sd'
fmap-ctx { · } unit _ = unit
fmap-ctx {Γ , A} (γ , a) p = fmap-ctx γ p , fmap-ty {A} a p
```

Functoral mapping for left-hand sides and simple right-hand sides

Similarly, we can use transitivity of order to define the functorial mapping of the denotational semantics of left-hand sides, as the definition of left-hand sides requires a proof. Since simple right-hand sides are defined from left-hand sides and literals

```
fmap-L : ∀ {sd sd'} → L sd → sd ≤ₛ sd' → L sd'
fmap-L (l-var sdᵛ sdᵛ≤ₛsd) sd≤ₛsd' = l-var sdᵛ (≤ₛ-trans sdᵛ≤ₛsd sd≤ₛsd')
fmap-L (l-sbrs) _ = l-sbrs

fmap-S : ∀ {sd sd'} → S sd → sd ≤ₛ sd' → S sd'
fmap-S (s-l l) sd≤ₛsd' = s-l (fmap-L l sd≤ₛsd')
fmap-S (s-lit lit) _ = s-lit lit
```

Functorial mapping for instruction sequences[5]

The functorial mapping of the denotational semantics of instruction sequences is defined as follows, assuming the stack descriptor $sd' = \langle f', d' \rangle$ is greater than $sd = \langle f, d \rangle$:

- When $f < f'$, we use the popto instruction that directly pops the stack to sd;
- When $f' = f$ and $d < d'$, we use adjustdisp to decrease the displacement by $d' - d$.

In the Agda implementation, we also need to show that after the the adjustment, the stack descriptor is now sd. This requires the term n−[n−m]≡m proved in §3.5.3. The functorial mapping of the denotational semantics of instruction sequences is defined as follows:

```
fmap-I : ∀ {sd sd'} → I sd → sd ≤s sd' → I sd'
fmap-I {sd} c (<-f f<f') = popto sd (<-f f<f') c
fmap-I {⟨ f , d ⟩} {⟨ f , d' ⟩} c (≤-d d≤d') =
  adjustdisp-dec ((d' − d) d≤d') (−→≤ d≤d')
    (I-sub {n = (d' − d) d≤d'} (n−[n−m]≡m d≤d') c)
```

3.6.5 Denotational semantics of terms

The denotational semantics of terms is defined as a morphism from the denotational semantics of contexts to the denotational semantics of types. Similarly, it is a family of continuous functions indexed by the stack descriptor. The denotational semantics of terms has the following type in Agda:

```
⟦_⟧ : ∀ {Γ A} → Γ ⊢ A → (sd : SD) → ⟦ Γ ⟧ctx sd → ⟦ A ⟧ty sd
```

We need to find out the denotational semantics of the terms corresponding to all rules in the typing judgement mentioned in §3.3.3.

Variables

We define an auxiliary function that simply gets the variable from the list of variables (the context), and the denotational semantics of variables is given by applying this function:

```
⟦_⟧var : ∀ {Γ A sd} → A ∈ Γ → ⟦ Γ ⟧ctx sd → ⟦ A ⟧ty sd
⟦ Zero ⟧var (_ , a) = a
⟦ Suc b ⟧var (γ , _) = ⟦ b ⟧var γ

⟦ Var a ⟧ sd γ = ⟦ a ⟧var γ
```

[5] In the work of Reynolds, the functorial mapping of instruction sequences is presented as the functorial mapping of Compl. The two definitions are equivalent as we define Compl $sd = \mathsf{I}_{sd}$. However, I think it is more natural to present it as the functorial mapping of instruction sequences, as the definition consists of instruction sequences.

Subtyping

The idea of subtyping is to use a term of type A as a term of type A', where A is a subtype of A'. We define an auxiliary function that takes in the subtype relation $A \leq: A'$ and the denotational semantics of A, and returns the denotational semantics of A'.

- For integer variables as subtypes of integer expressions and integer acceptors, since we defined the denotational semantics of integer variable as product of the denotational semantics of integer expression and integer acceptor in §3.6.3, I can simply use the projection to get the denotational semantics of either side.
- For reflexivity, we simply use the identity function.
- For transitivity, we use the composition of the two functions.
- For contravariant function subtyping, we use a contravariant function.

The denotational semantics of subtyping is given by applying the auxiliary function:

$$
\begin{aligned}
&[\![\_]\!]\mathsf{sub} : \forall\ \{A\ A'\ sd\} \to A \leq: A' \to [\![\ A\ ]\!]\mathsf{ty}\ sd \to [\![\ A'\ ]\!]\mathsf{ty}\ sd \\
&[\![\ \leq\text{:-refl}\ ]\!]\mathsf{sub}\ a = a \\
&[\![\ \leq\text{:-trans}\ A{\leq}{:}A'\ A'{\leq}{:}A''\ ]\!]\mathsf{sub}\ a = [\![\ A'{\leq}{:}A''\ ]\!]\mathsf{sub}\ ([\![\ A{\leq}{:}A'\ ]\!]\mathsf{sub}\ a) \\
&[\![\ \leq\text{:-fn}\ A{\leq}{:}A'\ B'{\leq}{:}B\ ]\!]\mathsf{sub}\ a = \\
&\quad \lambda\ sd{\leq_s}sd'\ a' \to [\![\ B'{\leq}{:}B\ ]\!]\mathsf{sub}\ (a\ sd{\leq_s}sd'\ ([\![\ A{\leq}{:}A'\ ]\!]\mathsf{sub}\ a')) \\
&[\![\ \text{var-}\leq\text{:-exp}\ ]\!]\mathsf{sub}\ (exp\ ,\ acc) = exp \\
&[\![\ \text{var-}\leq\text{:-acc}\ ]\!]\mathsf{sub}\ (exp\ ,\ acc) = acc
\end{aligned}
$$

$$
[\![\ \mathsf{Sub}\ a\ A{\leq}{:}B\ ]\!]\ sd\ \gamma = [\![\ A{\leq}{:}B\ ]\!]\mathsf{sub}\ ([\![\ a\ ]\!]\ sd\ \gamma)
$$

Lambda abstraction

- For lambda, we need to extend the context γ with the extra variable in the lambda term. This is done by using the functorial mapping of the denotational semantics of contexts.
- For application, we simply apply the denotational semantics of the lambda term to the denotational semantics of the argument, where the proof is simply the reflexivity of the order of stack descriptors.

The denotational semantics of lambda abstraction is defined as follows:

$$
\begin{aligned}
&[\![\ \mathsf{Lambda}\ f\ ]\!]\ sd\ \gamma\ \{sd' = sd'\}\ sd{\leq_s}sd'\ a \\
&\quad = [\![\ f\ ]\!]\ sd'\ (\mathsf{fmap\text{-}ctx}\ \gamma\ sd{\leq_s}sd'\ ,\ a)
\end{aligned}
$$

$$
[\![\ \mathsf{App}\ f\ e\ ]\!]\ sd\ \gamma = [\![\ f\ ]\!]\ sd\ \gamma\ (\leq\text{-d}\ \leq\text{-refl})\ ([\![\ e\ ]\!]\ sd\ \gamma)
$$

Commands

- For skip, we simply return the continuation κ without changing it.
- For sequence, we need to prefix the continuation with the denotational semantics of the second command, and then prefix it with the denotational semantics of the first command, as the continuation κ is to be performed after both commands.
- For assignment, the definition is similar to sequence, as we need to prefix the continuation with the denotational semantics of the integer accepter and then prefix it with the denotational semantics of the integer expression.

The denotational semantics of commands is defined as follows:

$$[\![\text{Skip}]\!]\ sd\ \gamma\ sd{\leq_s}sd'\ \kappa = \kappa$$

$$[\![\text{Seq}\ c_1\ c_2]\!]\ sd\ \gamma\ sd{\leq_s}sd'\ \kappa = [\![c_1]\!]\ sd\ \gamma\ sd{\leq_s}sd'\ ([\![c_2]\!]\ sd\ \gamma\ sd{\leq_s}sd'\ \kappa)$$

$$[\![\text{Assign}\ a\ e]\!]\ sd\ \gamma\ sd{\leq_s}sd'\ \kappa = [\![e]\!]\ sd\ \gamma\ sd{\leq_s}sd'\ ([\![a]\!]\ sd\ \gamma\ sd{\leq_s}sd'\ \kappa)$$

The denotational semantics of new variable is more complicated. We first place a new variable in the stack with the current stack descriptor sd' by assigning literal 0 with the current stack descriptor sd' and set the displacement adjustment to 1, which corresponds to the place taken by this new variable. Then we do the command in NewVar, deallocate the new variable and do the rest of the commands in the continuation by applying the denotational semantics of the command in NewVar to an adjustdisp instruction that adjusts the displacement by -1, followed by the continuation κ. Here the new stack descriptor used is $sd' +_s 1$. The new environment (denotational semantics of context) γ' is the old environment γ with the denotational semantics of the new variable added.

The denotational semantics of the new variable is a product of the denotational semantics of an integer expression and the denotational semantics of an integer acceptor by definition in §3.6.3.

- The expression component fills the a given integer continuation β with the stack descriptor for the new variable;
- The acceptor component prefixes the continuation κ with an assignment of the new variable.

Please see Appendix [TBD] for the full implementation of the function NewVar.

Integer expressions

- For integer literals, we simply fill in the integer literal (as an R term) to the integer continuation β.

- For plus and negation, we wish to fill the negation of the given expression or the sum of the two given expressions to the integer continuation β. However, this is only possible when the given expression(s) are simple left-hand sides, as the grammar of the target language in Figure 3.7 restricts right-hand sides to contain at most one operator. To solve this problem, we introduce an auxiliary function use-temp.

The auxiliary function use-temp checks if the given expression is a simple left-hand side,

- If it is, we simply fill the expression to the integer continuation β[6].
- If it is not, we need to use a temporary variable to store the result of the expression, and then fill the temporary variable to the integer continuation β.

However, regarding the stack descriptor, according to Reynolds' model, we assume

- sd is the stack descriptor passed along the environment γ, describing any program variables that may be accessed during the evaluation of the whole expression (plus/negation on the given expression).
- sd' is the current stack descriptor before the evaluation of the whole expression, which is guaranteed to be not less than sd. sd' is also the position of the temporary variable in the stack.
- sd'' is the current stack descriptor after the evaluation on the given expression and just before the assignment of the temporary variable. sd'' is guaranteed to be not less than sd'.
- $sd''' = sd' +_s 1$ is the stack descriptor after the assignment of the temporary variable, as it is just large enough to hold the temporary variable.

In Reynolds' model, we simply adjust the stack descriptor from sd'' to sd''' by the adjustment δ in the assignment instruction. However, in the Agda implementation, we need to know whether the adjustment is positive or negative, using assign-inc and assign-dec respectively. We also need to show that the adjustment is valid for each case. This is done by using the totality of order of natural numbers mentioned in §3.5.1 and case split. Please refer to Appendix [TBD] for the full implementation of the function use-temp.

Given the definition of use-temp, the denotational semantics of plus and negation is simply wrapping β with a lambda function using use-temp to get the temporary variable, and directly fill the plus/negation of the variable to the integer continuation β. Note that in the denotational semantics of plus, we need to use the functorial mapping of simple right-hand sides for the first argument, as the stack may grow during the evaluation of the second argument. The

[6]The type of β here is actually a bit different to integer continuation we defined, as it only accepts simple right-hand sides instead of right-hand sides. Reynolds gave a wrong definition of the type of this term. I have corrected it in my Agda implementation.

denotational semantics of integer expressions is defined as follows:

```
⟦ Lit i ⟧ sd γ sd≤ₛsd' κ = κ ≤ₛ-refl (r-s (s-lit i))

⟦ Neg e ⟧ sd γ sd≤ₛsd' κ =
  ⟦ e ⟧ sd γ sd≤ₛsd'
    (use-temp λ sd≤ₛsd' s → κ sd≤ₛsd' (r-unary UNeg s))

⟦ Plus e₁ e₂ ⟧ sd γ p κ =
  ⟦ e₁ ⟧ sd γ p (use-temp (λ p' s₁ → ⟦ e₂ ⟧ sd γ (≤ₛ-trans p p')
                  (use-temp (λ p'' s₂ → κ (≤ₛ-trans p' p'')
                    (r-binary (fmap-S s₁ p'') BPlus s₂)))))
```

3.6.6 Compilation

Compilation is a functor from the source language to the target language. To define the compilation functor, we use the denotational semantics of terms and fill in the initial stack descriptor ($\langle 0, 0 \rangle$), the empty context (unit), a trivial proof for stack descriptor being not less than itself ($\leq_s$-refl) and the continuation representing the last instruction (stop):

```
compile-closed : · ⊢ comm → I ⟨ 0 , 0 ⟩
compile-closed t = ⟦ t ⟧ ⟨ 0 , 0 ⟩ unit ≤ₛ-refl stop
```

Chapter 4

Evaluation

Contents

4.1 **Tests** 42
4.1.1 Test cases 42
4.1.2 Feature checklist 43
4.2 **Extensibility** 45
4.3 **Success criteria** 45

Chapter 3 describes the implementation of the source language, the target language and the compiler in Agda. This chapter evaluates the implementation with integration tests and a feature checklist and then discusses the success criteria of the project.

4.1 Tests

4.1.1 Test cases

The correctness of the compiler is evaluated with a set of test cases. Each test is defined as follows

```
term-0 : · ⊢ comm
term-0 = Skip -- source term
result-0 = compile-closed term-0

test-0 : result-0 ≡ stop -- target term
test-0 = refl
```

Figure 4.1: Test 0 as an example of test cases

where definition of the compile-closed function is specified in §3.6.6. The program type checks only if both of the source term and the target term are correctly defined, and the target term is the result of the compilation of the source term.

There are special tests to check that some expressions in the source language compiles to the same expression in the target language. Such test is defined as follows:

```
test-3' : result-3' ≡ result-3
test-3' = refl
```

Figure 4.2: Test 3' as an example for checking equivalent compiled results

The full test cases are available in the appendix [insert link]. Here is a summary of the test cases written in a more readable format:

Test 0. skip

Test 1. $x := 2$

Test 2. $x := (\lambda a.\ a)\ 4$

Test 3. $x := (\lambda a.\ (\lambda b.\ a + b)\ 2)\ 3$

Test 3'. $x := 3 + 2$

Test 4. $x := -3;\ y := (\lambda a.x)\ 2$

Test 5. $x := 2;\ x := x + 1$

Test 5'. $x := 2;$ skip; $x := x + 1$

Test 6. $x := 2;\ x := -x + 1$

Test 7. $x := (\lambda a.\ -a + 1)\ 2$

where **Test 3** and **Test 3'**, **Test 5** and **Test 5'** compile to the same target term.

Test 3 and **Test 3'** are equivalent in terms of compilation because the compilation directly substitutes the value in corresponding lambda terms.

Test 5 and **Test 5'** are equivalent in terms of compilation because skip is a no-op in the source language.

4.1.2 Feature checklist

Traditional unit tests are not suitable for this project, as many typing judgement rules cannot be tested in isolation. For example, testing subtyping rule (Figure 3.3) or any of the integer expression typing rules (Figure 3.6) inherently requires embedding them in a well-typed command (NewVar in Figure 3.5), as the compiler only accepts complete commands.

Instead, I employed integration testing with a feature checklist, ensuring comprehensive coverage by validating all typing rules and targeting edge cases.

Feature	Typing judgement	Specification	Test cases number
Variable	Var	-	1, 2, 3, 3', 4, 5, 5', 6, 7
Subtyping	Sub	intvar $\leq:$ intacc	1, 2, 3, 3', 4, 5, 5', 6, 7
		intvar $\leq:$ intexp	4, 5, 5', 6
Lambda abstraction	Lambda	-	2, 3, 4, 7
		Multiple variables*	3
		Free variables*	4
	App	-	2, 3, 4, 7
Command	Skip	-	0, 5'
	Seq	-	4, 5, 5', 6
		Multiple sequencing	5'
	NewVar	-	1, 2, 3, 3', 4, 5, 5', 6, 7
		Multiple variables*	4
	Assign	-	1, 2, 3, 3', 4, 5, 5', 6, 7
		Self-referential assignments	5, 5', 6
Integer expression	Lit	Natural number	1, 3, 3', 4, 5, 6, 7
		Negative number	2, 4
	Neg	-	5, 5', 6, 7
	Plus	-	3, 3', 6, 7
	-	Multiple expressions†	6, 7

Table 4.1: Feature checklist

Here are some notes on the feature checklist:

* These tests ensure that the translation of context as a whole is correct.

† This type of test is necessary because the target language does not support doing multiple integer operations in a single instruction, as specified in Figure 3.7. As a result, special tests are required to check that the compiler correctly translates integer expressions with multiple operations into a continuation of single operations defined in the target language.

The feature checklist includes all the typing rules in the source language, including edge cases including multiple variables, free variables and self-referential assignments. Additionally, I account for target-language-specific constraints, ensuring multiple operations are translated into simple instructions with continuation. The tests cover all the cases in the feature checklist, ensuring that the compiler generates expected target code under all circumstances.

4.2 Extensibility

This project is inherently extensible, as all of the components are defined as terms and types in Agda. The compiler can be extended to support more features by adding new terms in the form of type judgement rules in the source language, adding any new instructions in the target language if needed, and write corresponding compilation rules.

4.3 Success criteria

The project has met the success criteria as follows:

- A formalisation of the source language has been given in Agda, which includes the syntax, typing rules and operational semantics.
- A formalisation of the target language has been given in Agda, which includes all instructions from the instruction set. Rigorous definitions of the operation of the stack descriptors are given, ensuring correctness and preserving the mathematical properties of the stack descriptors.
- A formalisation of the compiler has been given in Agda, which includes the denotational semantics of the source language and a term for compilation.
- Integration tests have been given in Agda, which cover all the case in the feature checklist. The tests ensure that the compiler generates expected target code under all circumstances.

The following are the extension criteria that are met:

- A customised library have been developed, which includes rigorous definitions of subtraction of natural numbers and corresponding properties. Two approaches to rigorous definition of subtraction has been implemented and compared.
- Operational semantics of the source language has been defined with renaming, substitution and reduction rules.
- Complicated functions defined for translating expressions with multiple operations in the source language to a continuation of simple instructions in the target language.

Chapter 5

Conclusion

Contents

5.1 Results 46
5.2 Lessons learned 46
5.3 Future work 47

5.1 Results

This project has successfully implemented a compiler from a source language (STLC with store) to a target language (stack machine) in Agda. The compiler is defined as a functor from the source language to the target language, which is directly generated from the denotational semantics of the source language. The compiler validates and refines Reynolds' theory [6], offers a concrete example of how category-theoretic semantics can generate intermediate code.

5.2 Lessons learned

The project has been quite challenging as I have to learn both category theory and Agda from scratch, and apply them to implementation. The implementation process revealed unexpected gaps between Reynolds' theoretical framework and the computational realisation: while his denotational semantics provided core definitions, actualising them in Agda required formalising numbers, their operations, and properties and diagnosing and resolving type mismatches and ambiguities in the original paper [6]. These hurdles highlighted how even meticulously defined theories can be complicated when translated to working systems.

In terms of planning the project, the project has deviated slightly from the original timetable in the project proposal due to differences between the anticipated structure of the project and its actual development process. When I was drafting the proposal, I expected the workflow to be similar to a standard compiler, so I divided the work into components including basic instructions, variable declarations and conditionals, allocating two weeks for each. In hindsight, these components are similar term definitions, while the core of the project can be more accurately divided into three main tasks: defining the source language, defining the target language, and defining the translation between the two. The most time-consuming aspect of each

step has been understanding their formal definitions in the paper and addressing the missing details in the Agda implementation.

Overall, the project has been a valuable learning experience, and I have gained a deeper understanding of both Category Theory and Agda and an exciting glimpse of how the correspondences between logic, type theory and category theory can be used to implement a compiler.

5.3 Future work

The project has successfully implemented the compiler, but there are still many areas for improvement and further research:

- The compiler can be extended to support more features, such as subroutines and iterations.
- The operational semantics of the target language can be defined, which can be used, along with the operational semantics of the source language, to prove the correctness of the compiler.
- Reynolds pointed out that the requirement of instructions being natural transformation must be relaxed to allow for instruction sequences that differ operationally but are equivalent in terms of denotational semantics [6, Ch. 6]. Extra research can be done to develop a reasonable equational theory that can be stated directly on the instruction sequences, under which the semantics actually becomes a true functor category model (i.e. naturality holds).

Bibliography

[1] R. Loader, *Notes on simply typed lambda calculus*, 1996. [Online]. Available: https://www.lfcs.inf.ed.ac.uk/reports/98/ECS-LFCS-98-381/.

[2] J. Lambek, "Cartesian closed categories and typed lambda- calculi," in *Proceedings of the Thirteenth Spring School of the LITP on Combinators and Functional Programming Languages*, Berlin, Heidelberg: Springer-Verlag, 1985, pp. 136–175, ISBN: 3540171843.

[3] J. C. Reynolds, "The essence of algol," in *ALGOL-like Languages, Volume 1*. USA: Birkhauser Boston Inc., 1997, pp. 67–88, ISBN: 0817638806.

[4] F. J. Oles, "A category-theoretic approach to the semantics of programming languages," AAI8301650, Ph.D. dissertation, USA, 1982.

[5] F. J. Oles, "Type algebras, functor categories, and block structure," *DAIMI Report Series*, vol. 12, no. 156, Jan. 1983. DOI: 10.7146/dpb.v12i156.7430. [Online]. Available: https://tidsskrift.dk/daimipb/article/view/7430.

[6] J. C. Reynolds, "Using functor categories to generate intermediate code," in *Proceedings of the 22nd ACM SIGPLAN-SIGACT Symposium on Principles of Programming Languages*, ser. POPL '95, San Francisco, California, USA: Association for Computing Machinery, 1995, pp. 25–36, ISBN: 0897916921. DOI: 10.1145/199448.199452. [Online]. Available: https://doi.org/10.1145/199448.199452.

[7] U. Norell, "Dependently typed programming in agda," in *Proceedings of the 4th International Workshop on Types in Language Design and Implementation*, ser. TLDI '09, Savannah, GA, USA: Association for Computing Machinery, 2009, pp. 1–2, ISBN: 9781605584201. DOI: 10.1145/1481861.1481862. [Online]. Available: https://doi.org/10.1145/1481861.1481862.

[8] X. Leroy, "Formal verification of a realistic compiler," *Commun. ACM*, vol. 52, no. 7, pp. 107–115, Jul. 2009, ISSN: 0001-0782. DOI: 10.1145/1538788.1538814. [Online]. Available: https://doi.org/10.1145/1538788.1538814.

[9] R. Kumar, M. O. Myreen, M. Norrish, and S. Owens, "Cakeml: A verified implementation of ml," in *Proceedings of the 41st ACM SIGPLAN-SIGACT Symposium on Principles of Programming Languages*, ser. POPL '14, San Diego, California, USA: Association for Computing Machinery, 2014, pp. 179–191, ISBN: 9781450325448. DOI: 10.1145/2535838.2535841. [Online]. Available: https://doi.org/10.1145/2535838.2535841.

[10] S. Castellan, P. Clairambault, and P. Dybjer, *Categories with families: Unityped, simply typed, and dependently typed*, 2020. arXiv: 1904.00827 [cs.LO]. [Online]. Available: https://arxiv.org/abs/1904.00827.

[11] "Chapter 10 first order dependent type theory," in *Categorical logic and type theory*, ser. Studies in Logic and the Foundations of Mathematics, B. Jacobs, Ed., vol. 141, Elsevier, 1998, pp. 581–644. DOI: https://doi.org/10.1016/S0049-237X(98)80040-2. [Online]. Available: https://www.sciencedirect.com/science/article/pii/S0049237X98800402.

[12] J. Z. S. Hu and J. Carette, "Formalizing category theory in agda," in *Proceedings of the 10th ACM SIGPLAN International Conference on Certified Programs and Proofs*, ser. CPP 2021, Virtual, Denmark: Association for Computing Machinery, 2021, pp. 327–342, ISBN: 9781450382991. DOI: 10.1145/3437992.3439922. [Online]. Available: https://doi.org/10.1145/3437992.3439922.

[13] P. Wadler, W. Kokke, and J. G. Siek, *Programming Language Foundations in Agda*. Aug. 2022. [Online]. Available: https://plfa.inf.ed.ac.uk/20.08/.

[14] T. Leinster, *Basic category theory*, 2016. arXiv: 1612.09375 [math.CT]. [Online]. Available: https://arxiv.org/abs/1612.09375.

[15] D. S. Scott, "Relating theories of the λ-calculus," in *Relating Theories of the Lambda-Calculus: Dedicated to Professor H. B. Curry on the Occasion of His 80th Birthday*, Oxford: Springer, 1974, p. 406.

[16] E. Riehl, *Category Theory in Context*. Mineola, NY: Dover Publications, 2016. [Online]. Available: https://math.jhu.edu/~eriehl/context.pdf.

[17] A. Pitts and M. Fiore, *Category theory lecture notes*, Lecture notes, University of Cambridge, 2025. [Online]. Available: https://www.cl.cam.ac.uk/teaching/2425/CAT/CATLectureNotes.pdf.

[18] T. G. Griffin, "A formulae-as-type notion of control," in *Proceedings of the 17th ACM SIGPLAN-SIGACT Symposium on Principles of Programming Languages*, ser. POPL '90, San Francisco, California, USA: Association for Computing Machinery, 1989, pp. 47–58, ISBN: 0897913434. DOI: 10.1145/96709.96714. [Online]. Available: https://doi.org/10.1145/96709.96714.

[19] The Agda Community, *Agda standard library*, version 2.1.1, Release date: 2024-09-07, 2024. [Online]. Available: https://github.com/agda/agda-stdlib.

[20] *Git*, Accessed: 2025-04-15. [Online]. Available: https://git-scm.com/.

[21] *Github*, Accessed: 2025-04-15. [Online]. Available: https://github.com/.

[22] *Gnu emacs*, Accessed: 2025-04-15. [Online]. Available: https://www.gnu.org/software/emacs/.

[23] *Visual studio code*, Accessed: 2025-04-15. [Online]. Available: https://code.visualstudio.com/.

[24] T.-G. LUA, *Agda-mode*, Accessed: 2025-04-15. [Online]. Available: https://marketplace.visualstudio.com/items?itemName=banacorn.agda-mode.

[25] *Windows subsystem for linux (wsl)*, Accessed: 2025-04-15. [Online]. Available: https://ubuntu.com/desktop/wsl.

Appendix A

Operational Semantics of the source language

The operational semantics of the source language is defined with renaming, substitution and reduction rules. The implementation is as follows:

```
data Value : ∀{Γ A} → Γ ⊢ A → Set where
  V-Lambda : ∀{Γ A B} {F : Γ , A ⊢ B} → Value (Lambda {Γ} F)
  V-Lit : ∀{Γ} {i : ℤ} → Value (Lit {Γ} i)
  V-Skip : ∀{Γ} → Value (Skip {Γ})

-- Renaming
ext : ∀{Γ Δ} → (∀{A} → A ∈ Γ → A ∈ Δ)
             → (∀{A B} → B ∈ Γ , A → B ∈ Δ , A)
ext ρ Zero = Zero
ext ρ (Suc x) = Suc (ρ x)

rename : ∀{Γ Δ} → (∀{A} → A ∈ Γ → A ∈ Δ)
                → (∀{A} → Γ ⊢ A → Δ ⊢ A)
rename ρ (Var A∈Γ) = Var (ρ A∈Γ)
rename ρ (Lambda Γ,A⊢B) = Lambda (rename (ext ρ) Γ,A⊢B)
rename ρ (Sub Γ⊢A A≤:B) = Sub (rename ρ Γ⊢A) A≤:B
rename ρ (App Γ⊢A Γ⊢B) = App (rename ρ Γ⊢A) (rename ρ Γ⊢B)
rename ρ Skip = Skip
rename ρ (Seq Γ⊢c₁ Γ⊢c₂) = Seq (rename ρ Γ⊢c₁) (rename ρ Γ⊢c₂)
rename ρ (NewVar Γ⊢c) = NewVar (rename (ext ρ) Γ⊢c)
rename ρ (Assign Γ⊢i Γ⊢e) = Assign (rename ρ Γ⊢i) (rename ρ Γ⊢e)
rename ρ (Lit Γ⊢i) = Lit Γ⊢i
rename ρ (Neg Γ⊢i) = Neg (rename ρ Γ⊢i)
rename ρ (Plus Γ⊢i₁ Γ⊢i₂) = Plus (rename ρ Γ⊢i₁) (rename ρ Γ⊢i₂)
```

Figure A.1: Operational semantics of the source language (Part 1)

```
-- Simultaneous substitution
exts : ∀{Γ Δ} → (∀{A} → A ∈ Γ → Δ ⊢ A)
               → (∀{A B} → B ∈ Γ , A → Δ , A ⊢ B)
exts σ Zero = Var Zero
exts σ (Suc x) = rename Suc (σ x)

subst : ∀{Γ Δ} → (∀{A} → A ∈ Γ → Δ ⊢ A)
                → (∀{A} → Γ ⊢ A → Δ ⊢ A)
subst σ (Var A∈Γ) = σ A∈Γ
subst σ (Sub Γ⊢A A≤:B) = Sub (subst σ Γ⊢A) A≤:B
subst σ (Lambda Γ,A⊢B) = Lambda (subst (exts σ) Γ,A⊢B)
subst σ (App Γ⊢A Γ⊢B) = App (subst σ Γ⊢A) (subst σ Γ⊢B)
subst σ Skip = Skip
subst σ (Seq Γ⊢c₁ Γ⊢c₂) = Seq (subst σ Γ⊢c₁) (subst σ Γ⊢c₂)
subst σ (NewVar Γ⊢c) = NewVar (subst (exts σ) Γ⊢c)
subst σ (Assign Γ⊢i Γ⊢e) = Assign (subst σ Γ⊢i) (subst σ Γ⊢e)
subst σ (Lit Γ⊢i) = Lit Γ⊢i
subst σ (Neg Γ⊢i) = Neg (subst σ Γ⊢i)
subst σ (Plus Γ⊢i₁ Γ⊢i₂) = Plus (subst σ Γ⊢i₁) (subst σ Γ⊢i₂)

-- Single substitution
_[_] : ∀{Γ A B} → Γ , B ⊢ A → Γ ⊢ B → Γ ⊢ A
_[_] {Γ} {A} {B} N M = subst {Γ , B} {Γ} σ {A} N
  where
  σ : ∀ {A} → A ∈ Γ , B → Γ ⊢ A
  σ Zero = M
  σ (Suc x) = Var x

-- Reduction
data _⟶_ : ∀{Γ A} → (Γ ⊢ A) → (Γ ⊢ A) → Set where
  App-cong₁ : ∀{Γ A B} {F F' : Γ ⊢ A ⇒ B} {E : Γ ⊢ A}
                    → F ⟶ F' → App F E ⟶ App F' E
  App-cong₂ : ∀{Γ A B} {V : Γ ⊢ A ⇒ B} {E E' : Γ ⊢ A}
                    → Value V → E ⟶ E' → App V E ⟶ App V E'
  Lambda-β : ∀{Γ A B} {F : Γ , A ⊢ B} {V : Γ ⊢ A}
                    → Value V → App (Lambda F) V ⟶ F [ V ]
```

Figure A.2: Operational semantics of the source language (Part 2)

Using Functor Categories to Generate Intermediate Code in Agda

Computer Science Tripos Part II Project Proposal

Jack Gao (yg410), Homerton College

Project Supervisors: Yulong Huang (yh419) and Yufeng Li (yl959)

Director of Studies: Dr. John Fawcett

October 2024

Introduction

In the paper "Using Functor Categories to Generate Intermediate Code"[1], John Reynold proposed a way of translating Algol-like languages (declarative programming languages with stores) to intermediate code, using their semantic interpretations in functor categories. By picking a suitable category of intermediate code, the interpretation process becomes compilation, which is "correct by construction". Reynold concluded that he did not have a proper dependently typed language in hand, so his compiler is a partial function.

With the development of dependently typed languages and proof-based languages, we can implement his compiler as a total function and prove compiler correctness. In this project, I will use Agda to implement the compiler. Agda captures the source language's intrinsic syntax with indexed families, which contains only well-typed terms. Therefore, it focuses on compiling the correct programs and rules out the ill-typed nonsensical input. The correctness of the compiler is supported by both theory and implementation. The theory of functor category semantics is based on the denotational semantics model, which offers a mathematically rigorous framework of understanding the meaning of programs. As a proof assistant language, Agda provides formal proofs with its strong type system, so the correctness of the compiler can be proved along its implementation. This also follows the increasing trend in formally proving the correctness of compilers and having verified compilers.

The project is to implement the compiler as described in [1] in Agda. The source language is an Algol-like language and the target language is an assembly-style intermediate language for a stack machine. The core part is compiling the basic instructions, variable declarations, and conditionals. John Reynold's paper provides a detailed and precise definition of these components, which makes the implementation of the compiler very feasible.

Possible extensions include compiling advanced features such as open calls, subroutines and iterations, optimising the target code, writing proofs of correctness of the compiler, and exploring the equational theory of instruction sequences using the functor category semantics as a guide, which was not presented in the paper. Reynolds' claims to build a functor category model, but at the end of Chapter 6 in [1] admits that he has actually done no such thing as the naturality laws do not hold unless instruction sequences are replaced by their image in some denotational model that is left unspecified. The extension could be exploring a reasonable equational theory that can be stated directly on the instruction sequences, under which the semantics actually becomes a true functor category model (i.e. naturality holds).

Substance and Structure

Theoretical foundation: functor category semantics based on the denotational semantics model

My implementation:
- Compiler: written in Agda, a dependent typed proof-based language
- Source language: an Algol-like language (a declarative programming language with stores)
- Target language: an assembly-style intermediate language for a stack machine

The specification of the source language and the target language will be given in the project.

Starting Point

I have not learned Agda prior to the project. I am aware that there is an online open-source tutorial for Agda[2]. In preparation for the project, I set up the environment for Agda on my laptop according to "Front Matter" in the book, but did not read anything from Part 1.

I do not have any other experience with compiler beyond Part IB Compiler Construction Course. I have not learned category theory and type theory prior to the lectures in Part II.

Evaluation

The current approach involves only type-checking the Agda program without executing it, as the performance is expected to be highly inefficient. At present, no effective solution to this issue has been proposed in academia. Consequently, evaluating the compiler's performance would not provide relevant insights.

Rigorously proving the correctness of a compiler involves intricate proofs rooted in equational theory. I propose implementing a series of unit tests on the generated intermediate code. The correctness can be demonstrated by comparing the output of the target code with the expected results.

Success Criteria

The project will be considered successful if the following conditions are met:
- A specification of an Algol-like language is given
- The Agda code for compiler successfully compiles given the specified Algol-like language
- A specification of an assembly-style intermediate language written
- The compiler output an language that satisfies the specification
- Basic instructions, variable declarations and conditionals can be compiled

Work Plan

Dates	Deliverable
24 Oct–8 Jan	Completion of Core Objectives
9 Jan–22 Jan	Progress Report and Presentation
23 Jan–5 Mar	Extensions, Proofs and Tests
6 Mar–14 May	Dissertation

Work packages breakdown:

1. **Michaelmas weeks 3–4**: 24 Oct–6 Nov
   - Learn Agda and be familiar with its different usage.

   Milestone: completing the exercises on Programming Language Foundations in Agda
2. **Michaelmas weeks 5–6**: 7 Nov–20 Nov
   - Implement basic instructions of the compiler.

   Milestone: code for basic instructions completed.
3. **Michaelmas weeks 7–8**: 21 Nov–4 Dec
   - Implement variable declarations of the compiler.

   Milestone: code for variable declarations completed.
4. **Christmas weeks 1–2**: 5 Dec–18 Dec

   Slack period:
   - Finish implementing basic instructions and variable declarations of the compiler.
   - Write proofs for basic instructions and variable declarations of the compiler if possible.

   Milestone: code for basic instructions and variable declarations completed

   Possible Milestone: proofs for basic instructions and variable declarations written.
5. **Christmas weeks 3–4**: 19 Dec–1 Jan
   - Implement conditionals of the compiler.

   Milestone: code for conditionals completed.
6. **Christmas weeks 5**: 2 Jan–8 Jan

   Slack period:
   - Check all core part functions.
   - Write unit tests for the generated intermediate code.
   - Write proofs for conditionals if possible.
   - Implement iterations (extension) and write proofs if possible.

Milestone: code for basic instructions, variable declarations and conditionals completed; all corresponding proofs written; unit tests written.

Possible Milestone: code for iterations completed; proofs for conditionals and iterations written.

7. **Christmas week 6–7**: 9 Jan–22 Jan

   - Write a progress report.
   - Prepare presentation slides.

   Milestone: progress report written; presentation slides prepared.

8. **Lent weeks 1–2**: 23 Jan–5 Feb

   - Rehearse the presentation.
   - Write unit tests for the generated intermediate code.
   - Check the feasibility of extensions.
   - Work on extensions.

   *Milestone: progress report submitted (**Due 7 Feb**); presentation prepared and rehearsed with supervisor; partial result of extensions; unit tests written.*

9. **Lent weeks 3–4**: 6 Feb–19 Feb

   - Give presentation.
   - Work on extensions.
   - Show correctness of translation by proving it satisfies the necessary properties.

   Milestone: presentation given; partial result of extensions.

   Possible Milestone: Proofs for extensions written.

10. **Lent weeks 5–6**: 20 Feb–5 Mar

    Slack period:
    - Complete the extensions.
    - Finish all proofs of the compiler.
    - Finish all unit tests of the generated intermediate code.

    Milestone: extensions completed; all unit tests written.

    Possible Milestone: Proofs of all components of the compiler written.

11. **Lent weeks 7–8**: 6 Mar–19 Mar

    - Draft introduction and preparation chapter of the dissertation and get feedback from supervisors.

    Milestone: introduction and preparation chapter written and checked.

12. **Easter vacation weeks 1–2**: 20 Mar–2 Apr

    - Draft implementation chapter and get feedback from supervisors.

    Milestone: implementation chapter written and checked.

13. **Easter vacation weeks 3–4**: 3 Apr–16 Apr

    - Draft evaluation and conclusion chapter and get feedback from supervisors.

    Milestone: evaluation and conclusion chapter written and checked.

14. **Easter vacation weeks 5–6**: 17 Apr–30 Apr

    Slack period:
    - adjust dissertation based on feedback from supervisors.

    Milestone: dissertation completed.

15. **Easter weeks 1–2**: 1 May–14 May

    - Final proof-reading and submit dissertation.

    *Milestone: dissertation and source code submitted. (**Due 16 May**)*

Resource Declaration

I will use my personal laptop for this project. My laptop specifications are:
- Lenovo Legion Y7000P IRH8
- CPU: 13th Gen Intel(R) Core(TM) i7-13700H 2.40GHz
- RAM: 16 GB
- SSD: 1TB + 2TB
- OS: Windows 11 Home

Windows Subsystem for Linux is installed on the machine. The version is Ubuntu 22.04 LTS.

Should this machine fail, I will continue my work on my other laptop. I will use Git for version control and daily work will push to a GitHub repository.

The project requires an Agda compiler, which is open-source and freely available online. I have already installed and tested it on my laptop.

References

[1] J. C. Reynolds, "Using Functor Categories to Generate Intermediate Code.," in *Proceedings of the 22nd ACM SIGPLAN-SIGACT Symposium on Principles of Programming Languages*, Association for Computing Machinery, 1995. doi: 10.1145/199448.199452.

[2] Philip Wadler, Wen Kokke, and Jeremy G. Siek, *Programming Language Foundations in Agda*. 2022. [Online]. Available: https://plfa.inf.ed.ac.uk/22.08/