```agda
+-identityʳ : ∀ {n} → n + zero ≡ n
+-identityʳ {zero} = refl
+-identityʳ {suc n} rewrite +-identityʳ {n} = refl

+-sucʳ : ∀ {m n} → m + suc n ≡ suc (m + n)
+-sucʳ {zero} {n} = refl
+-sucʳ {suc m} {n} rewrite +-sucʳ {m} {n} = refl

+-comm : ∀ {m n} → m + n ≡ n + m
+-comm {m} {zero} = +-identityʳ
+-comm {m} {suc n}
  rewrite (+-sucʳ {m} {n}) | (+-comm {m} {n}) = refl



≤-irrelevant : ∀ {m n} → (p₁ p₂ : m ≤ n) → p₁ ≡ p₂
≤-irrelevant z≤n z≤n = refl
≤-irrelevant (s≤s p₁) (s≤s p₂) = cong s≤s (≤-irrelevant p₁ p₂)

≤-refl : ∀ {n : ℕ} → n ≤ n
≤-refl {zero} = z≤n
≤-refl {suc n} = s≤s ≤-refl

≤-trans : ∀ {m n p : ℕ} → m ≤ n → n ≤ p → m ≤ p
≤-trans z≤n _ = z≤n
≤-trans (s≤s m≤n) (s≤s n≤p) = s≤s (≤-trans m≤n n≤p)

n≤suc-n : ∀ {n : ℕ} → n ≤ suc n
n≤suc-n {zero} = z≤n
n≤suc-n {suc n} = s≤s n≤suc-n

m≡n,p≤n→p≤m : ∀ {p m n} → m ≡ n → p ≤ n → p ≤ m
m≡n,p≤n→p≤m m≡n p≤n rewrite sym m≡n = p≤n

+→≤ : ∀ {m n : ℕ} → m ≤ m + n
+→≤ {zero} {n} = z≤n
+→≤ {suc m} {n} = s≤s +→≤

+→≤ʳ : ∀ {m n : ℕ} → m ≤ n + m
+→≤ʳ {m} {n} = m≡n,p≤n→p≤m (+-comm {n} {m}) +→≤

data Order : ℕ → ℕ → Set where
  leq : ∀ {m n : ℕ} → m ≤ n → Order m n
  geq : ∀ {m n : ℕ} → n ≤ m → Order m n

≤-compare : ∀ {m n : ℕ} → Order m n
≤-compare {zero} {n} = leq z≤n
≤-compare {suc m} {zero} = geq z≤n
≤-compare {suc m} {suc n} with ≤-compare {m} {n}
... | leq m≤n = leq (s≤s m≤n)
... | geq n≤m = geq (s≤s n≤m)

data _<_ : ℕ → ℕ → Set where
  z<s : ∀ {n : ℕ} → zero < suc n
  s<s : ∀ {m n : ℕ} → m < n → suc m < suc n

<→s≤ : ∀ {m n : ℕ} → m < n → suc m ≤ n
<→s≤ (z<s) = s≤s z≤n
<→s≤ (s<s m<n) = s≤s (<→s≤ m<n)

<→≤ : ∀ {m n : ℕ} → m < n → m ≤ n
<→≤ m<n = ≤-trans n≤suc-n (<→s≤ m<n)

<-trans : ∀ {m n p : ℕ} → m < n → n < p → m < p
<-trans z<s (s<s _) = z<s
<-trans (s<s m<n) (s<s n<p) = s<s (<-trans m<n n<p)

_−_ : (n : ℕ) → (m : ℕ) → (p : m ≤ n) → ℕ
(n − zero) (z≤n) = n
(suc n − suc m) (s≤s m≤n) = (n − m) m≤n

−-irrelevant : ∀ {n m} → (p₁ p₂ : m ≤ n) → (n − m) p₁ ≡ (n − m) p₂
−-irrelevant {n} {m} p₁ p₂ rewrite ≤-irrelevant p₁ p₂ = refl

−→≤ : ∀ {n m} → (m≤n : m ≤ n) → (n − m) m≤n ≤ n
−→≤ z≤n = ≤-refl
−→≤ (s≤s m≤n) = ≤-trans (−→≤ m≤n) n≤suc-n

n−n≡0 : ∀ {n} → (n − n) (≤-refl {n}) ≡ 0
n−n≡0 {zero} = refl
n−n≡0 {suc n} = n−n≡0 {n}

−-suc : ∀ {n m} → {m≤n : m ≤ n}
          → suc ((n − m) m≤n) ≡ (suc n − m) (≤-trans m≤n n≤suc-n)
−-suc {_} {zero} {z≤n} = refl
−-suc {suc n} {suc m} {s≤s m≤n} = −-suc {n} {m} {m≤n}

n−[n−m]≡m : ∀ {m n} → (m≤n : m ≤ n)
                → (n − ((n − m) m≤n)) (−→≤ m≤n) ≡ m
n−[n−m]≡m {zero} {n} z≤n = n−n≡0 {n}
n−[n−m]≡m {suc m} {suc n} (s≤s m≤n) =
  trans (sym (−-suc {n} {(n − m) m≤n}))
          (cong suc (n−[n−m]≡m {m} {n} m≤n))

n+m−m≡n : ∀ {m n} → (n + m − m) (+→≤ʳ) ≡ n
n+m−m≡n {m} {zero} =
  trans (−-irrelevant {m} {m} +→≤ʳ ≤-refl) (n−n≡0 {m})
n+m−m≡n {m} {suc n} =
  trans
    (−-irrelevant {suc n + m} {m} +→≤ʳ (≤-trans +→≤ʳ n≤suc-n))
    (trans (sym (−-suc {n + m} {m}))
                (cong suc (n+m−m≡n {m} {n})))

suc-d≤d′→d≤d′−[d′−[suc-d]] : ∀ {d d′} → (δ₁≤δ₂ : suc d ≤ d′)
          → d ≤ ((d′ − ((d′ − (suc d)) δ₁≤δ₂)) (−→≤ δ₁≤δ₂))
suc-d≤d′→d≤d′−[d′−[suc-d]] δ₁≤δ₂ =
          m≡n,p≤n→p≤m (n−[n−m]≡m δ₁≤δ₂) n≤suc-n
```